
Overview
    

Absinthe is the GraphQL toolkit for Elixir, an implementation of the GraphQL specification built to suit the language's capabilities and idiomatic style.
The Absinthe project consists of several complementary packages. You can find the full listing on the absinthe-graphql GitHub organization page.

  
    
  
  GraphQL Basics


If you're new to GraphQL, we suggest you read up a bit on GraphQL's foundational principles before you dive into Absinthe.
Here are a few resources that might be helpful:
	The official GraphQL website
	How to GraphQL, which includes a brief tutorial using Absinthe


  
    
  
  Absinthe


Absinthe's functionality generally falls into two broad areas. You can read more about the details in the guides provided as part of this documentation and in the related packages/projects:
	Defining Schemas. A schema:	defines the structure of data entities and the relationships between, as well as the available queries, mutations, and subscriptions, using an elegant collection of declarative macros
	defines custom scalar types
	declares any deprecated definitions
	defines resolution functions to access data, using a flexible and extensible middleware/plugin system


	Executing Documents. A GraphQL document:	can be any standard GraphQL query, mutation, or subscription
	may include reusable variable definitions
	can be analyzed for its complexity and be rejected if it's unsafe/too expensive
	has a context that you can integrate with authentication and authorization strategies
	can contain standard GraphQL introspection fields
	can include multipart file uploads as GraphQL arguments (as part of the absinthe_plug package)




  
    
  
  Integrations


Absinthe integrates with a number of other important projects, both on the backend and frontend, to provide a better experience for developers.
	Elixir	Support for HTTP APIs using Plug and Phoenix via the absinthe_plug and absinthe_phoenix packages
	Support for Ecto via the dataloader package


	JavaScript (client-side)	Support for Relay and Apollo Client
	Support for Absinthe's channel-based subscriptions. See absinthe-socket.




  
    
  
  Guides


To contribute to the guides, please submit a pull request to the absinthe project on GitHub.
You'll find the content under guides/.


  

    
Installation
    

To install Absinthe, just add an entry to your mix.exs:
def deps do
  [
    # ...
    {:absinthe, "~> 1.7"}
  ]
end
(Check Hex to make sure you're using an up-to-date version number.)

  
    
  
  Overriding Dependencies


Because the Absinthe project is made up of a large number of related packages to support integrations with other tools, sometimes you may want to update only part of your absinthe-related dependencies.
Don't forget you can use the :override option for your Mix dependencies if you'd like to ensure a specific package is at a specific version number. For example, If you wanted to try a new version of Absinthe without updating something that depends on it (which is locked to an older version):
def deps do
  [
    # ...
    {:absinthe, "~> 1.7", override: true}
  ]
end

  
    
  
  Plug, Phoenix, and GraphiQL


Most people use Absinthe to support an HTTP API.
You'll want to read the Plug and Phoenix for specific installation and configuration options, including how you can run the handy, included GraphiQL tool directly from your application.


  

    
Learning
    

The following are some Absinthe-specific educational resources that are available.

  
    
  
  Books


	Craft GraphQL APIs in Elixir with Absinthe by the creators of Absinthe.


  
    
  
  Online Resources


	Website (mostly just links elsewhere)
	Documentation (current stable release)
	How to GraphQL (with Absinthe)


  
    
  
  Videos


	Live APIs with GraphQL Subscriptions, ElixirConf 2017 (Bellevue)
	GraphQL in Practice, ElixirConf EU 2017 (Barcelona)


  
    
  
  General GraphQL Information


There's a ton of GraphQL resources on the web.
The official website and How to GraphQL are good places to start.


  

    
Community
    


  
    
  
  Twitter


Follow the project on Twitter as @absinthegraphql for news and additional resources.

  
    
  
  Chat


You can find the maintainers and an active community of users and contributors in the #absinthe-graphql channel in the Elixir Slack.

  
    
  
  Forum


Questions and suggestions can be submitted on the Elixir Forum. Please categorize/tag as Absinthe.


  

    
Getting Started
    

We'll be building a very basic GraphQL API for a blog, written in Elixir using
Absinthe.

  
    
  
  Background


Before you start, it's a good idea to have some background into GraphQL in general. Here are a few resources that might be helpful:
	The official GraphQL website
	How to GraphQL (this includes another brief tutorial using Absinthe)


  
    
  
  The Example


 The tutorial expects you to have a properly set-up Phoenix application with absinthe and absinthe_plug added to the dependencies.
 If you'd like to cheat, you can find the finished code for the tutorial
 in the Absinthe Example
 project on GitHub.


  
    
  
  First Step


Let's get started with our first query!


  

    
Our First Query
    

The first thing our viewers want is a list of our blog posts, so
that's what we're going to give them. Here's the query we want to
support:
{
  posts {
    title
    body
  }
}
To do this we're going to need a schema. Let's create some basic types
for our schema, starting with a :post. GraphQL has several fundamental
types on top of which all of our types will be
built. The Absinthe.Type.Object type is the right one
to use when representing a set of key value pairs.
Since our Post Ecto schema lives in the Blog.Content Phoenix
context, we'll define its GraphQL counterpart type, :post, in a
matching BlogWeb.Schema.ContentTypes module:
In blog_web/schema/content_types.ex:
defmodule BlogWeb.Schema.ContentTypes do
  use Absinthe.Schema.Notation

  object :post do
    field :id, :id
    field :title, :string
    field :body, :string
  end
end
The GraphQL specification requires that type names be unique, TitleCased words.
Absinthe does this automatically for us, extrapolating from our type identifier
(in this case :post gives us "Post"). If really needed, we could provide a
custom type name as a :name option to the object macro.

If you're curious what the type :id is used by the :id field, see
the GraphQL spec. It's
an opaque value, and in our case is just the regular Ecto id, but
serialized as a string.
With our type completed we can now write a basic schema that will let
us query a set of posts.
In blog_web/schema.ex:
defmodule BlogWeb.Schema do
  use Absinthe.Schema
  import_types BlogWeb.Schema.ContentTypes

  alias BlogWeb.Resolvers

  query do

    @desc "Get all posts"
    field :posts, list_of(:post) do
      resolve &Resolvers.Content.list_posts/3
    end

  end

end
More information
For more information on the macros available to build a schema, see
their definitions in Absinthe.Schema and Absinthe.Schema.Notation.
This uses a resolver module we've created (again, to match the Phoenix context naming)
at blog_web/resolvers/content.ex:
defmodule BlogWeb.Resolvers.Content do

  def list_posts(_parent, _args, _resolution) do
    {:ok, Blog.Content.list_posts()}
  end

end
Queries are defined as fields inside the GraphQL object returned by
our query function. We created a posts query that has a type
list_of(:post) and is resolved by our
BlogWeb.Resolvers.Content.list_posts/3 function. Later we'll talk
more about the resolver function parameters; for now just remember
that resolver functions can take two forms:
	A function with an arity of 3 (taking a parent, arguments, and resolution struct)
	An alternate, short form with an arity of 2 (omitting the first parameter, the parent)

The job of the resolver function is to return the data for the
requested field. Our resolver calls out to the Blog.Content module,
which is where all the domain logic for posts lives, invoking its
list_posts/0 function, then returns the posts in an :ok tuple.
Note
Resolvers can return a wide variety of results, to include errors and configuration
for advanced plugins that further process the data.
If you're asking yourself what the implementation of the domain logic looks like, and exactly how
the related Ecto schemas are built, read through the code in the absinthe_tutorial
repository. The tutorial content here is intentionally focused on the Absinthe-specific code.
Now that we have the functional pieces in place, let's configure our
Phoenix router to wire this into HTTP:
In blog_web/router.ex:
defmodule BlogWeb.Router do
  use BlogWeb, :router

  pipeline :api do
    plug :accepts, ["json"]
  end

  scope "/api" do
    pipe_through :api

    forward "/graphiql", Absinthe.Plug.GraphiQL,
      schema: BlogWeb.Schema

    forward "/", Absinthe.Plug,
      schema: BlogWeb.Schema

  end

end
In addition to our API, we've wired in a handy GraphiQL user interface to play with it. Absinthe integrates both the classic GraphiQL and more advanced GraphiQL Workspace interfaces as part of the absinthe_plug package.
Now let's check to make sure everything is working. Start the server:
$ mix phx.server

Absinthe does a number of sanity checks during compilation, so if you misspell a type or make another schema-related gaffe, you'll be notified.
Once it's up-and-running, take a look at http://localhost:4000/api/graphiql:
[image: ]Make sure that the URL is pointing to the correct place and press the play button. If everything goes according to plan, you should see something like this:
[image: ]
  
    
  
  Next Step


Now let's look at how we can add arguments to our queries.


  

    
Query Arguments
    

Our GraphQL API would be pretty boring (and useless) if clients
couldn't retrieve filtered data.
Let's assume that our API needs to add the ability to look-up users by
their ID and get the posts that they've authored. Here's what a basic query to do that
might look like:
{
  user(id: "1") {
    name
    posts {
      id
      title
    }
  }
}
The query includes a field argument, id, contained within the
parentheses after the user field name. To make this all work, we need to modify
our schema a bit.

  
    
  
  Defining Arguments


First, let's create a :user type and define its relationship to
:post while we're at it. We'll create a new module for the
account-related types and put it there; in
blog_web/schema/account_types.ex:
defmodule BlogWeb.Schema.AccountTypes do
  use Absinthe.Schema.Notation

  @desc "A user of the blog"
  object :user do
    field :id, :id
    field :name, :string
    field :email, :string
    field :posts, list_of(:post)
  end

end
The :posts field points to a list of :post results. (This matches
up with what we have on the Ecto side, where Blog.Accounts.User
defines a has_many association with Blog.Content.Post.)
We've already defined the :post type, but let's go ahead and add an
:author field that points back to our :user type. In
blog_web/schema/content_types.ex:
object :post do

  # post fields we defined earlier...

  field :author, :user

end
Now let's add the :user field to our query root object in our
schema, defining a mandatory :id argument and using the
Resolvers.Accounts.find_user/3 resolver function. We also need to
make sure we import the types from BlogWeb.Schema.AccountTypes so
that :user is available.
In blog_web/schema.ex:
defmodule BlogWeb.Schema do
  use Absinthe.Schema

  import_types Absinthe.Type.Custom

  # Add this `import_types`:
  import_types BlogWeb.Schema.AccountTypes

  import_types BlogWeb.Schema.ContentTypes

  alias BlogWeb.Resolvers

  query do

    @desc "Get all posts"
    field :posts, list_of(:post) do
      resolve &Resolvers.Content.list_posts/3
    end

    # Add this field:
    @desc "Get a user of the blog"
    field :user, :user do
      arg :id, non_null(:id)
      resolve &Resolvers.Accounts.find_user/3
    end

  end

end
Now let's use the argument in our resolver. In blog_web/resolvers/accounts.ex:
defmodule BlogWeb.Resolvers.Accounts do

  def find_user(_parent, %{id: id}, _resolution) do
    case Blog.Accounts.find_user(id) do
      nil ->
        {:error, "User ID #{id} not found"}
      user ->
        {:ok, user}
    end
  end

end
Our schema marks the :id argument as non_null, so we can be
certain we will receive it. If :id is left out of the query,
Absinthe will return an informative error to the user, and the resolve
function will not be called.
If you have experience writing Phoenix controller actions, you might
wonder why we can match incoming arguments with atoms instead of
having to use strings.
The answer is simple: you've defined the arguments in the schema
using atom identifiers, so Absinthe knows what arguments will be
used ahead of time, and will coerce as appropriate---culling any
extraneous arguments given to a query. This means that all arguments
can be supplied to the resolve functions with atom keys.

Finally you'll see that we can handle the possibility that the query,
while valid from GraphQL's perspective, may still ask for a user that
does not exist. We've decided to return an error in that case.
There's a valid argument for just returning {:ok, nil} when a
record can't be found. Whether the absence of data constitutes an
error is a decision you get to make.


  
    
  
  Arguments and Non-Root Fields


Let's assume we want to query all posts from a user published within a
given time range. First, let's add a new field to our :post object
type, :published_at.
The GraphQL specification doesn't define any official date or time
types, but it does support custom scalar types (you can read more
about them in the related guide), and
Absinthe ships with several built-in scalar types. We'll use
:naive_datetime (which doesn't include timezone information) here.
Edit blog_web/schema/content_types.ex:
defmodule BlogWeb.Schema.ContentTypes do
  use Absinthe.Schema.Notation

  @desc "A blog post"
  object :post do
    field :id, :id
    field :title, :string
    field :body, :string
    field :author, :user
    # Add this:
    field :published_at, :naive_datetime
  end
end
To make the :naive_datetime type available, add an import_types line to
your blog_web/schema.ex:
import_types Absinthe.Type.Custom
For more information about how types are imported,
read the guide on the topic.
For now, just remember that import_types should only be
used in top-level schema module. (Think of it like a manifest.)

Here's the query we'd like to be able to use, getting the posts for a user
on a given date:
{
  user(id: "1") {
    name
    posts(date: "2017-01-01") {
      title
      body
      publishedAt
    }
  }
}
To use the passed date, we need to update our :user object type and
make some changes to its :posts field; it needs to support a :date
argument and use a custom resolver. In blog_web/schema/account_types.ex:
defmodule BlogWeb.Schema.AccountTypes do
  use Absinthe.Schema.Notation

  alias BlogWeb.Resolvers

  object :user do
    field :id, :id
    field :name, :string
    field :email, :string
    # Add the block here:
    field :posts, list_of(:post) do
      arg :date, :date
      resolve &Resolvers.Content.list_posts/3
    end
  end

end
For the resolver, we've added another function head to
Resolvers.Content.list_posts/3. This illustrates how you can use the
first argument to a resolver to match the parent object of a field. In
this case, that parent object would be a Blog.Accounts.User Ecto
schema:
# Add this:
def list_posts(%Blog.Accounts.User{} = author, args, _resolution) do
  {:ok, Blog.Content.list_posts(author, args)}
end
# Before this:
def list_posts(_parent, _args, _resolution) do
  {:ok, Blog.Content.list_posts()}
end
Here we pass on the user and arguments to the domain logic function,
Blog.Content.list_posts/3, which will find the posts for the user
and date (if it's provided; the :date argument is optional). The
resolver, just as when it's used for the top level query :posts,
returns the posts in an :ok tuple.
Tip
Check out the full implementation of logic for
Blog.Content.list_posts/3--and some simple seed data--in
the
absinthe_tutorial repository.
If you've done everything correctly (and have some data handy), if you
start up your server with mix phx.server and head over
to http://localhost:4000/api/graphiql, you should be able to play
with the query.
It should look something like this:
[image: ]
  
    
  
  Next Step


Next up, we look at how to modify our data using mutations.


  

    
Mutations
    

A blog is no good without new content. We want to support a mutation
to create a blog post:
mutation CreatePost {
  createPost(title: "Second", body: "We're off to a great start!") {
    id
  }
}
Now we just need to define a mutation portion of our schema and
a :create_post field:
In blog_web/schema.ex:
mutation do

  @desc "Create a post"
  field :create_post, type: :post do
    arg :title, non_null(:string)
    arg :body, non_null(:string)
    arg :published_at, :naive_datetime

    resolve &Resolvers.Content.create_post/3
  end

end
The resolver in this case is responsible for making any changes and
returning an {:ok, post} tuple matching the :post type we defined
earlier:
In our blog_web/resolvers/content.ex module, we'll add the
create_post/3 resolver function:
def create_post(_parent, args, %{context: %{current_user: user}}) do
  Blog.Content.create_post(user, args)
end
def create_post(_parent, _args, _resolution) do
  {:error, "Access denied"}
end
Obviously things can go wrong in a mutation. To learn more about the
types of error results that Absinthe supports, read the guide.


  
    
  
  Authorization


This resolver adds a new concept: authorization. The resolution struct
(that is, an Absinthe.Resolution)
passed to the resolver as the third argument carries along with it the
Absinthe context, a data structure that serves as the integration
point with external mechanisms---like a Plug that authenticates the
current user. You can learn more about how the context can be used in
the Context and Authentication
guide.
Going back to the resolver code:
	If the match for a current user is successful, the underlying
Blog.Content.create_post/2 function is invoked. It will return a
tuple suitable for return. (To read the Ecto-related nitty-gritty,
check out the absinthe_tutorial
repository.)
	If the match for a current user isn't successful, the fall-through
match will return an error indicating that a post can't be created.


  
    
  
  Next Step


Now let's take a look at more complex arguments


  

    
Complex Arguments
    

In preparation for supporting comments on our blog, let's create users. We're building
a modern mobile first blog of course, and thus want to support either a phone number
or an email as the contact method for a user.
We want to support the following mutations.
Support creation of a user with their email address:
mutation CreateEmailUser {
  createUser(
    contact: { type: EMAIL, value: "foo@bar.com" }
    name: "Jane"
    password: "hunter1"
  ) {
    id
    contacts {
      type
      value
    }
  }
}
And by using their phone number:
mutation CreatePhoneUser {
  createUser(
    contact: { type: PHONE, value: "+1 123 5551212" }
    name: "Joe"
    password: "hunter2"
  ) {
    id
    contacts {
      type
      value
    }
  }
}
To do this we need the ability to create nested arguments. GraphQL has input objects
for this purpose. Input objects, like regular object, contain key value pairs, but
they are intended for input only (you can't do circular references with them for example).
Another notion we'll look at here is an enumerable type. We only want to support contact
types "email" and "phone" at the moment, and GraphQL gives us the ability to
specify this in our schema.
Let's start with our :contact_type Enum. In blog_web/schema/account_types.ex:
enum :contact_type do
  value :phone, as: "phone"
  value :email, as: "email"
end
We're using the :as option here to make sure the parsed enum is represented by a string
when it's passed to our controllers; this is to ease integration with our Ecto schema
(by default, the enum values are passed as atoms).
Note
The standard convention for representing incoming enum values in
GraphQL documents are in all caps. For instance, given our settings
here, the accepted values would be PHONE and EMAIL (without
quotes). See the GraphQL document examples above for examples.
While the enum macro supports configuring this incoming format, we
highly recommend you just use the GraphQL convention.
Now if a user tries to send some other kind of contact type they'll
get a nice error without any extra effort on your part. Enum types are
not a substitute for modeling layer validations however, be sure to
still enforce things like this on that layer too.
Now for our contact input object.
In blog_web/schema/account_types.ex:
input_object :contact_input do
  field :type, non_null(:contact_type)
  field :value, non_null(:string)
end
Note that we name this type :contact_input. Input object types have
their own names, and the _input suffix is common.
Important
It's very important to remember that only input
types---basically scalars and input objects---can be used to model
input.
Finally our schema, in blog_web/schema.ex:
mutation do

  #... other mutations

  @desc "Create a user"
  field :create_user, :user do
    arg :name, non_null(:string)
    arg :contact, non_null(:contact_input)
    arg :password, non_null(:string)

    resolve &Resolvers.Accounts.create_user/3
  end

end
Suppose in our database that we store contact information in a different database
table. Our mutation would be used to create both records in this case.
There does not need to be a one to one correspondence between how data is structured
in your underlying data store and how things are presented by your GraphQL API.
Our resolver, blog_web/resolvers/accounts.ex might look something like this:
def create_user(_parent, args, %{context: %{current_user: %{admin: true}}}) do
  Blog.Accounts.create_user(args)
end
def create_user(_parent, args, _resolution) do
  {:error, "Access denied"}
end
You'll notice we're checking for :current_user again in our Absinthe
context, just as we did before for posts. In this case we're taking
the authorization check a step further and verifying that only
administrators (in this simple example, an administrator is a user
account with :admin set to true) can create a user.
Everyone else gets an "Access denied" error for this field.
Note
To see the Ecto-related implementation of the
Blog.Accounts.create_user/1 function and the (stubbed) authentication logic we're
using for this example, see the absinthe_tutorial
repository.
Here's our mutation in action in GraphiQL.
[image: ]Note
Note we're sending a Authorization header to authenticate, which a
plug is handling. Make sure to read the
related guide for more
information on how to set-up authentication in your own
applications.
Our simple tutorial application is just using a simple stub: any
authorization token logs you in the first user. Obviously not what
you want in production!

  
    
  
  Next Step


Now let's wrap things up.


  

    
Conclusion
    

With this we have a basic GraphQL based API for a blog. Head on over
to the github page if
you want the final code.
We hope to expand this tutorial to include a comment system that will
acquaint you with Union types and Fragments in the coming days.
Head on over to the topic guides for further reading, and see
the community page for information
on how to get help, ask questions, or contribute!

  
    
  
  Please Help!


This tutorial is a work in progress, and while it covers the basics of
using Absinthe, there is plenty more that can be added and improved
upon. It's important that it's kept up-to-date, too, so if you notice
something that's slipped by us, please help us fix it!
Please contribute your GitHub issues (and pull requests!):
	The tutorial text is under guides/tutorial in the absinthe
repository. It's in Markdown and easy to edit!
	The tutorial code located in the absinthe_tutorial repository.



  

    
Writing Schemas
    

A GraphQL API starts by building a schema. Using Absinthe, schemas are normal
modules that use Absinthe.Schema.
Here's a schema that supports looking up an item by ID:
# filename: myapp/schema.ex
defmodule MyAppWeb.Schema do

  use Absinthe.Schema

  # Example data
  @items %{
    "foo" => %{id: "foo", name: "Foo"},
    "bar" => %{id: "bar", name: "Bar"}
  }

  query do
    field :item, :item do
      arg :id, non_null(:id)
      resolve fn %{id: item_id}, _ ->
        {:ok, @items[item_id]}
      end
    end
  end

end
You may want to refer to the Absinthe API documentation for more detailed information as you look this over.
Some macros and functions used here that are worth mentioning, pulled in automatically from
Absinthe.Schema.Notation by use Absinthe.Schema:
	query - Defines the root query object. It's like using object but with
 nice defaults. There is a matching mutation macro as well.
	field - Defines a field in the enclosing object, input_object, or interface.
	arg - Defines an argument in the enclosing field or directive.
	resolve - Sets the resolve function for the enclosing field.

You'll notice we mention some types being referenced: :item and :id. :id
is a built-in scalar type (like :string, :boolean, and others), but :item
we need to define ourselves.
We can do it in the same MyAppWeb.Schema module, using the object macro defined by Absinthe.Schema.Notation:
# filename: myapp/schema.ex
@desc "An item"
object :item do
  field :id, :id
  field :name, :string
end
Now you can use Absinthe to execute a query document. Keep in mind that for
HTTP, you'll probably want to use
Absinthe.Plug instead of executing
GraphQL query documents yourself. Absinthe doesn't know or care about HTTP,
but the absinthe_plug project does: it handles the vagaries of interacting
with HTTP GraphQL clients so you don't have to.
If you were executing query documents yourself (let's assume for a local tool),
it would go something like this:
"""
{
  item(id: "foo") {
    name
  }
}
"""
|> Absinthe.run(MyAppWeb.Schema)

# Result
{:ok, %{data: %{"item" => %{"name" => "Foo"}}}}
Your schemas can be further customized using the options available to
Absinthe.Schema.Notation.field/4 to help provide for a richer experience for
your users, customize the field names, or mark fields as deprecated.
# filename: myapp/language_schema.ex
@desc "A Language"
object :language do
  field :id, :id
  field :iso_639_1, :string, description: "2 character ISO 639-1 code", name: "iso639"
  field :name, :string, description: "English name of the language"
end

  
    
  
  Importing Types


We could also move our type definitions out into a different module, for instance, MyAppWeb.Schema.Types, and then use import_types in our MyAppWeb.Schema:
# filename: myapp/schema/types.ex
defmodule MyAppWeb.Schema.Types do
  use Absinthe.Schema.Notation

  object :item do
    field :id, :id
    field :name, :string
  end

  # ...

end

# filename: myapp/schema.ex
defmodule MyAppWeb.Schema do
  use Absinthe.Schema

  import_types MyAppWeb.Schema.Types

  # ...

end
It's a nice way of separating the top-level query and mutation information,
which define the surface area of the API, with the actual types that it uses.
See Importing Types for a full guide to importing types.


  

    
Plug and Phoenix Setup
    

First, install Absinthe.Plug and a JSON codec of your choice,
eg, Jason:
# filename: mix.exs
def deps do
  [
    {:absinthe_plug, "~> 1.5"},
    {:jason, "~> 1.0"},
  ]
end

  
    
  
  Plug


To use, simply plug Absinthe.Plug in your pipeline.
plug Absinthe.Plug,
  schema: MyAppWeb.Schema
If you are going to support content types other than simply application/graphql
you should plug Absinthe.Plug after Plug.Parsers.
plug Plug.Parsers,
  parsers: [:urlencoded, :multipart, :json, Absinthe.Plug.Parser],
  json_decoder: Jason

plug Absinthe.Plug,
  schema: MyAppWeb.Schema
For more information on how the content types work, see General Usage.

  
    
  
  Phoenix


If your entire API is going to be based on GraphQL, we recommend simply plugging
Absinthe.Plug in at the bottom of your endpoint, and removing your router altogether.
defmodule MyApp.Endpoint do
  use Phoenix.Endpoint, otp_app: :my_app

  plug Plug.RequestId
  plug Plug.Logger

  plug Plug.Parsers,
    parsers: [:urlencoded, :multipart, :json],
    pass: ["*/*"],
    json_decoder: Jason

  plug Absinthe.Plug,
    schema: MyAppWeb.Schema
end
If you want only Absinthe.Plug to serve a particular route, configure your router
like:
defmodule MyAppWeb.Router do
  use Phoenix.Router

  resource "/pages", MyAppWeb.PagesController

  forward "/api", Absinthe.Plug,
    schema: MyAppWeb.Schema
end
Now Absinthe.Plug will only serve GraphQL from the /api url.

  
    
  
  Absinthe Context


Absinthe.Plug will pass any values found inside conn.private[:absinthe][:context]
on to Absinthe.run as the context. This is how you should handle logic that
uses headers -- most notably, Authentication.
For more information, see the Context guide.

  
    
  
  GraphiQL


See the absinthe_plug
project and the GraphiQL portion of the Introspection guide to
learn how to use the built-in Absinthe.Plug.GraphiQL plug.

  
    
  
  General Usage


This plug supports requests in a number of ways:

  
    
  
  Via a GET


With a query string:
?query=query+GetItem($id:ID!){item(id:$id){name}}&variables={id:"foo"}
Due to varying limits on the maximum size of URLs,
we recommend using one of the POST options below instead, putting the query into the body of the request.

  
    
  
  Via an application/json POST


With a POST body:
{
  "query": "query GetItem($id: ID!) { item(id: $id) { name } }",
  "variables": {
    "id": "foo"
  }
}
(We could also pull either query or variables out to the query string, just
as in the GET example.)

  
    
  
  Via an application/graphql POST


With a query string:
?variables={id:"foo"}
And a POST body:
query GetItem($id: ID!) {
  item(id: $id) {
    name
  }
}

  
    
  
  HTTP API


How clients interact with the plug over HTTP is designed to closely match that
of the official
express-graphql middleware.
In the example above, we went over the various ways to
make a request, but here are the details:
Once installed at a path, the plug will accept requests with the
following parameters:
	query - A string GraphQL document to be executed.

	variables - The runtime values to use for any GraphQL query variables
as a JSON object.

	operationName - If the provided query contains multiple named
operations, this specifies which operation should be executed. If not
provided, a 400 error will be returned if the query contains multiple
named operations.


The plug will first look for each parameter in the query string, eg:
/graphql?query=query+getUser($id:ID){user(id:$id){name}}&variables={"id":"4"}
If not found in the query string, it will look in the POST request body, using
a strategy based on the Content-Type header.
For content types application/json and application/x-www-form-urlencoded,
configure Plug.Parsers (or equivalent) to parse the request body before Absinthe.Plug, eg:
plug Plug.Parsers,
  parsers: [:urlencoded, :multipart, :json],
  pass: ["*/*"],
  json_decoder: Jason
For application/graphql, the POST body will be parsed as GraphQL query string,
which provides the query parameter. If variables or operationName are
needed, they should be passed as part of the

  
    
  
  Configuration Notes


As a plug, Absinthe.Plug requires very little configuration. If you want to support
application/x-www-form-urlencoded or application/json you'll need to plug
Plug.Parsers first.
plug Plug.Parsers,
  parsers: [:urlencoded, :multipart, :json],
  pass: ["*/*"],
  json_decoder: Jason

plug Absinthe.Plug,
  schema: MyApp.Linen.Schema
Absinthe.Plug requires a schema: config.
It also takes several options. See the documentation
for the full listing.

  
    
  
  Inside Phoenix controllers


You can use GraphQL as the datasource for your Phoenix controllers. For this 
you'll need to add absinthe_phoenix to your dependencies. See Absinthe Phoenix for installation instructions.
@graphql """
  query ($filter: UserFilter) {
    users(filter: $filter, limit: 10)
  }
"""
def index(conn, %{data: data}) do
  render conn, "index.html", data
end
The results of the query are now available in the "index.html" template. For
more information, see Absinthe.Phoenix.Controller


  

    
Writing Middleware and Plugins
    

Middleware enables custom resolution behaviour on a field. You can use them to share common logic that needs to happen before or after resolving fields. Things like authentication and error handling.

  
    
  
  Create a Middleware


In order to create a Middleware you need a module that implements Absinthe.Middleware behaviour . Your module needs to have a call/2 method which receives an %Absinthe.Resolution{} struct and some options as its parameters, and then returns a possibly altered resolution struct.
Here is an example of a middleware that handle Ecto.Changeset errors and makes sure we properly add error message to the errors in the response.
defmodule MyApp.Middlewares.HandleChangesetErrors do
  @behaviour Absinthe.Middleware
  def call(resolution, _) do
    %{resolution |
      errors: Enum.flat_map(resolution.errors, &handle_error/1)
    }
  end

  defp handle_error(%Ecto.Changeset{} = changeset) do
    changeset
      |> Ecto.Changeset.traverse_errors(fn {err, _opts} -> err end)
      |> Enum.map(fn({k,v}) -> "#{k}: #{v}" end)
  end
  defp handle_error(error), do: [error]
end
The resolution struct has all kinds of useful values inside of it. You can access the Absinthe context, the root value, information about the current field's AST, and more. For more information on how the current user ends up in the context please see our full authentication guide.

  
    
  
  Using Middlewares


Middleware can be placed on a field in few different ways:

  
    
  
  1. Using the Absinthe.Schema.Notation.middleware/2 macro used inside a field definition


This option is good when you want to add your middleware on few specific fields. You can use middleware to add multiple Middlewares before or after resolve. In this example MyApp.Web.Authentication would run before resolution, and HandleError would run after.
field :hello, :string do
  middleware MyApp.Web.Authentication
  resolve &get_the_string/2
  middleware HandleError, :foo
end

  
    
  
  2. Using the middleware/3 callback in your schema.


This option is good if you want to add your middleware on all or a group of fields based on the type of query. middleware/3 is a function callback on a schema. When you use Absinthe.Schema a default implementation of this function is placed in your schema. It is passed the existing middleware for a field, the field itself, and the object that the field is a part of.
You can override this callback to add your Middleware in the list of existing middlewares. In this example we add our HandleChangesetError Middleware only to mutations.
# add this to your schema module

# if it's a field for the mutation object, add this middleware to the end
def middleware(middleware, _field, %{identifier: :mutation}) do
  middleware ++ [MyApp.Middlewares.HandleChangesetErrors]
end
# if it's any other object keep things as is
def middleware(middleware, _field, _object), do: middleware

  
    
  
  3. Returning a {:middleware, middleware_spec, config} tuple from a resolution function.


You can update your resolution function to return {:middleware, MyApp.Middlewares.HandleChangesetErrors, config}, note that in this case the middleware can only be applied after the resolution.


  

    
Returning Errors
    

This guide could use some improvement.
You can help! Please fork the absinthe repository, edit guides/errors.md, and submit a pull request.
One or more errors for a field can be returned in a single {:error, error_value} tuple.
error_value can be:
	A simple error message string.
	A map containing :message key, plus any additional serializable metadata.
	A keyword list containing a :message key, plus any additional serializable metadata.
	A list containing multiple of any/all of these.
	Any other value compatible with to_string/1.


  
    
  
  Basic Errors


A simple error message:
{:error, "Something bad happened"}
Multiple error messages:
{:error, ["Something bad", "Even worse"]}
Single custom errors (note the required :message keys):
{:error, message: "Unknown user", code: 21}
{:error, %{message: "A database error occurred", details: format_db_error(some_value)}}
Three errors of mixed types:
{:error, ["Simple message", [message: "A keyword list error", code: 1], %{message: "A map error"}]}
Generic handler for interoperability with errors from other libraries:
{:error, :foo}
{:error, 1.0}
{:error, 2}

  
    
  
  Ecto.Changeset Errors


You may want to look at the Absinthe ErrorPayload package.


  

    
Batching Resolution
    


  
    
  
  Avoiding N+1 Queries


In general, you want to make sure that when accessing Ecto associations that you
preload the data in the top level resolver functions to avoid N+1 queries.
Imagine this scenario: You have posts and users. A Post has an author field, which
returns a user. You want to list all posts, and get the name of their author:
{
  posts {
    author {
      name
    }
  }
}
If you write your schema like this, you're going to have a bad time due to issues with N + 1:
object :post do
  @desc "Author of the post"
  field :author, :user do
    resolve fn post, _, _ ->
      author =
        post
        |> Ecto.assoc(:author)
        |> Repo.one

      {:ok, author}
    end
  end
end

query do
  field :posts, list_of(:post) do
    resolve fn _, _ ->
      {:ok, Post |> Repo.all}
    end
  end
end
What this schema will do when presented with the GraphQL query is
run Post |> Repo.all, which will retrieve N posts. Then for each
post it will resolve child fields, which runs our Repo.one query
function, resulting in N+1 calls to the database.
One way to handle this issue is with Absinthe's support for
batching. The idea with batching is that we're gonna aggregate all the
author_ids from each post, and then make one call to the user.
Let's first make a function to get a model by ids:
defmodule MyAppWeb.Schema.Helpers do
  def by_id(model, ids) do
    import Ecto.Query

    ids = ids |> Enum.uniq

    model
    |> where([m], m.id in ^ids)
    |> Repo.all
    |> Map.new(&{&1.id, &1})
  end
end
Now we can use this function to batch our author lookups:
object :post do

  @desc "Author of the post"
  field :author, :user do
    resolve fn post, _, _ ->
      batch({MyAppWeb.Schema.Helpers, :by_id, User}, post.author_id, fn batch_results ->
        {:ok, Map.get(batch_results, post.author_id)}
      end)
    end
  end

end
Now we make just two calls to the database. The first call loads all of the posts.
Then as Absinthe walks through each post and tries to get the author, it's instead
told to aggregate its information. That aggregate information is passed on to our by_id/2 function from earlier.
It grabs ALL the users in just one database call, and creates a map of user ids
to users.
Absinthe then does a second pass and calls the batch_results function with that
map, letting us retrieve the individual author for each post.
Not only is this a very efficient way to query the data, it's also 100% dynamic.
If a query document asks for authors, they're loaded efficiently. If it does not,
they aren't loaded at all.
See the documentation for Absinthe.Middleware.Batch for more information.
Absinthe.Middleware.Batch achieves a lot and, with some helpers, was the
standard way to solve this problem for a long time. While batching still has a
place, it has a few limitations that have driven the development of Dataloader.
There are small scale annoyances like the limitation of only being able to batch
one thing at a time in a field, or the fact that the API can get very verbose.
There's also some larger scale issues however. Ecto has a fair number of quirks
that make it a difficult library to abstract access to. If you want the
concurrent test system to work, you need to add self() to all the batch keys
and do Repo.all(caller: pid) in every batch function so that it knows which
sandbox to use. It gets very easy for your GraphQL functions to become full of
direct database access, inevitably going around important data access rules you
may want to enforce in your contexts. Alternatively, your context functions can
end up with dozens of little functions that only exist to support batching items
by ID.
In time, people involved in larger projects have been able to build some
abstractions, helpers, and conventions around the Absinthe.Middleware.Batch
plugin that have done a good job of addressing these issues. That effort has been
extracted into the Dataloader project, which also draws inspiration from similar
projects in the GraphQL world.
We've made it easier and more flexible, however, with
Elixir's dataloader package.

  
    
  
  Dataloader


Let's jump straight in to getting Dataloader working, and then we'll expand on
what's actually happening behind the scenes.
Using Dataloader is as simple as doing:
alias MyApp.Blog # Dataloader source, see below
import Absinthe.Resolution.Helpers, only: [dataloader: 1]

object :post do
  field :posts, list_of(:post), resolve: dataloader(Blog)

  @desc "Author of the post"
  field :author, :user do
    resolve dataloader(Blog)
  end
end
To make this work we need to setup a dataloader, add the Blog source to it, and
make sure our schema knows it needs to run the dataloader.
Latest install instructions found here: https://github.com/absinthe-graphql/dataloader
Let's start with a data source. Dataloader data sources are just structs that encode
a way of retrieving data in batches. In a Phoenix application you'll generally have one
source per context, so that each context can control how its data is loaded.
Here is a hypothetical Blog context and a dataloader ecto source:
defmodule MyApp.Blog do
  def data() do
    Dataloader.Ecto.new(MyApp.Repo, query: &query/2)
  end

  def query(queryable, _params) do
    queryable
  end
end
In this example, the query returned by query/2 is used as a starting point by Dataloader to build the final query, which it does by traversing schema associations.
In other words, Dataloader can determine that an author has many posts, and that to retrieve posts it needs to get those with the relevant author_id.
If that's sufficient for your needs, query/2 need not modify the query it's given.
But if you only want to load published posts, query/2 can narrow the query accordingly.
When integrating Dataloader with GraphQL, we want to place it in our context so
that we can access it in our resolvers. In your schema module add:
alias MyApp.{Blog}

def context(ctx) do
  loader =
    Dataloader.new
    |> Dataloader.add_source(Blog, Blog.data())

  Map.put(ctx, :loader, loader)
end

def plugins do
  [Absinthe.Middleware.Dataloader] ++ Absinthe.Plugin.defaults()
end
That's it! If you run a GraphQL query that hits that field, it will be loaded efficiently without N+1.
See the documentation for Dataloader for more information.


  

    
Dataloader
    

Dataloader provides an easy way efficiently load data in batches.
It's inspired by https://github.com/facebook/dataloader

  
    
  
  Installation


def deps do
  [
    {:dataloader, "~> 1.0.0"}
  ]
end

  
    
  
  Usage


The core concept of dataloader is a data source which is just a struct
that encodes a way of retrieving data. More info in the Sources section.

  
    
  
  Schema


Absinthe provides some dataloader helpers out of the box that you can import into your schema
import Absinthe.Resolution.Helpers, only: [dataloader: 1]
This is needed to use the various dataloader helpers to resolve a field:
field(:posts, list_of(:post), resolve: dataloader(Blog))
Let's start with a data source. Dataloader data sources are just structs that encode
a way of retrieving data in batches. In a Phoenix application you'll generally have one
source per context, so that each context can control how its data is loaded.
Here is a hypothetical Blog context and a dataloader ecto source:
defmodule MyApp.Blog do
  def data() do
    Dataloader.Ecto.new(MyApp.Repo, query: &query/2)
  end

  def query(queryable, _params) do
    queryable
  end
end
In this example, the query returned by query/2 is used as a starting point by Dataloader to build the final query, which it does by traversing schema associations.
In other words, Dataloader can determine that an author has many posts, and that to retrieve posts it needs to get those with the relevant author_id.
If that's sufficient for your needs, query/2 need not modify the query it's given.
But if you only want to load published posts, query/2 can narrow the query accordingly.
When integrating Dataloader with GraphQL, we want to place it in our context so
that we can access it in our resolvers. In your schema module add:
alias MyApp.{Blog, Foo}

def context(ctx) do
  loader =
    Dataloader.new
    |> Dataloader.add_source(Blog, Blog.data())
    |> Dataloader.add_source(Foo, Foo.data()) # Foo source could be a Redis source

  Map.put(ctx, :loader, loader)
end

def plugins do
  [Absinthe.Middleware.Dataloader] ++ Absinthe.Plugin.defaults()
end
The context/1 function is a callback specified by the Absinthe.Schema behaviour that gives
the schema itself an opportunity to set some values in the context that it may need in order to run.
The plugins/0 function has been around for a while, and specifies what plugins the schema needs to resolve.
See the documentation for more.
Unpacking Dataloader
The data/0 function creates an Ecto data source, to which you pass your repo and a query function. This query function
is called every time you want to load something, and provides an opportunity to apply arguments or
set defaults. So for example if you always want to only load non-deleted posts you can do:
def query(Post, _), do: from p in Post, where: is_nil(p.deleted_at)

def query(queryable, _), do: queryable
Now any time you're loading posts, you'll just get posts that haven't been
deleted.
We can also use the context to ensure access conditions, so we can only show deleted posts for admins:
def query(Post, %{has_admin_rights: true}), do: Post

def query(Post, _), do: from p in Post, where: is_nil(p.deleted_at)

def query(queryable, _), do: queryable
Helpfully, those rules are defined within your context, helping ensure
that it has the final say about data access.

  
    
  
  Sources


Dataloader ships with two different built in sources:
	Dataloader.Ecto - for easily pulling out data with ecto
	Dataloader.KV - a simple KV key value source.

KV
Here is a simple example of a loader using the KV source in combination with absinthe:
defmodule MyProject.Loaders.Nhl do
  @teams [%{
    id: 1,
    name: "New Jersey Devils",
    abbreviation: "NJD"
  },
  %{
    id: 2,
    name: "New York Islanders",
    abbreviation: "NYI"
  }
  # etc.
  ]

  def data() do
    Dataloader.KV.new(&fetch/2)
  end

  def fetch(:teams, [%{}]) do
    %{
      %{} => @teams
    }
  end

  def fetch(:team, args) do
   # must return a map keyed by the args
   # args is a list of the args used to resolve your field
   # for example, if you have arg(:foo, non_null(:string))
   # args will look like: [%{foo: "value of foo here")}]

    args
    |> Enum.reduce(%{}, fn(%{id: id} = arg, result) ->
      Map.put(result, arg, find_team(id))
    end)
  end

  def fetch(_batch, args) do
    args |> Enum.reduce(%{}, fn(arg, accum) -> Map.put(accum, arg, nil) end)
  end

  defp find_team(id) do
    @teams |> Enum.find(fn(t) -> t |> Map.get(:id) == id end)
  end
end
Dataloader.KV requires a load function that accepts a batch and args. It must return a map of values keyed by the args.
This is the purpose of the fetch/2 function. The dataloader/1 helper we imported above uses the field name as the batch, and a map where the argument name is the key. For example: fetch(:team, [%{ id: 1 }])
Pattern matching can be used to fetch differently depending on the batch. For example, when the :teams batch is requested, the args will actually be an empty map (i.e. %{}).
If you’re interested in more generic use of Dataloader, see the dataloader project source.


  

    
The Context and Authentication
    

Absinthe context exists to provide shared values to a given document execution.
A common use would be to pass in the current user of a given request. The context
is set at the call to Absinthe.run, and cannot be modified over the course of
a given execution.

  
    
  
  Basic Usage


As a basic example let's think about a profile page, where we want the current user
to be able to access basic information about themselves, but not other users.
First we'll need a very basic schema:
defmodule MyAppWeb.Schema do
  use Absinthe.Schema

  @fakedb %{
    "1" => %{name: "Bob", email: "bubba@foo.com"},
    "2" => %{name: "Fred", email: "fredmeister@foo.com"},
  }

  query do
    field :profile, :user do
      resolve fn _, _, _ ->
        # How could we get a current user here?
      end
    end
  end

  object :user do
    field :id, :id
    field :name, :string
    field :email, :string
  end
end
A query we might want could look like:
{
  profile {
    email
  }
}
If we're signed in as user 1, we should get only user 1's email, for example:
{
  "profile": {
    "email": "bubba@foo.com"
  }
}
In order to set the context, our call to Absinthe.run/3 should look like:
Absinthe.run(document, MyAppWeb.Schema, context: %{current_user: %{id: "1"}})
To access this, we need to update our query's resolve function:
query do
  field :profile, :user do
    resolve fn _, _, %{context: %{current_user: current_user}} ->
      {:ok, Map.get(@fakedb, current_user.id)}
    end
  end
end
And now it works!

  
    
  
  Context and Plugs


When using Absinthe.Plug you don't have direct access to the Absinthe.run call.
Instead, we can use Absinthe.Plug.put_options/2 to set context values which
Absinthe.Plug will use to pass it along to Absinthe.run.
Setting up your GraphQL context is as simple as writing a plug that inserts the
appropriate values into the connection.
Let's use this mechanism to set our current_user from the previous example via
an authentication header. We will use the same Schema as before.
First, our plug. We'll be checking the connection for the authorization header, and calling
out to some unspecified authentication mechanism.
defmodule MyAppWeb.Context do
  @behaviour Plug

  import Plug.Conn
  import Ecto.Query, only: [where: 2]

  alias MyApp.{Repo, User}

  def init(opts), do: opts

  def call(conn, _) do
    context = build_context(conn)
    Absinthe.Plug.put_options(conn, context: context)
  end

  @doc """
  Return the current user context based on the authorization header
  """
  def build_context(conn) do
    with ["Bearer " <> token] <- get_req_header(conn, "authorization"),
    {:ok, current_user} <- authorize(token) do
      %{current_user: current_user}
    else
      _ -> %{}
    end
  end

  defp authorize(token) do
    User
    |> where(token: ^token)
    |> Repo.one
    |> case do
      nil -> {:error, "invalid authorization token"}
      user -> {:ok, user}
    end
  end

end
This plug will use the authorization header to lookup the current user. If one
is found, it correctly sets the absinthe context. If you're using Guardian or
some other library that provides utilities for authenticating users you can use
those here too, and just add their output to the context.
If there is no current user it's better to simply not have the :current_user
key inside the map, instead of doing %{current_user: nil}. This way you can
just pattern match for %{current_user: user} in your code and not need to
worry about the nil case.
Using this plug is very simple. If we're just in a normal plug context we can
just make sure it's plugged prior to Absinthe.Plug
plug MyAppWeb.Context

plug Absinthe.Plug,
  schema: MyAppWeb.Schema
If you're using a Phoenix router, add the context plug to a pipeline.
defmodule MyAppWeb.Router do
  use Phoenix.Router

  resource "/pages", MyAppWeb.PagesController

  pipeline :graphql do
    plug MyAppWeb.Context
  end

  scope "/api" do
    pipe_through :graphql

    forward "/", Absinthe.Plug,
      schema: MyAppWeb.Schema
  end
end


  

    
Understanding Subscriptions
    

GraphQL subscriptions are a way to have events in the server push data out to clients in real time. The client submits a subscription document that asks for particular data, and then when events happen that document is run against that event and the resulting data is pushed out.
Like queries and mutations, subscriptions are not intrinsically tied to any particular transport, and they're built within Absinthe itself to be able to operate on many different platforms.
At the moment however the most common and fully featured platform that you can run them on with Elixir is via Phoenix channels, so this guide will walk you through the basics of getting them hooked up to a phoenix application.

  
    
  
  Absinthe.Phoenix Setup


Packages you'll need:
{:absinthe, "~> 1.7"},
{:phoenix_pubsub, "~> 2.0"},
{:absinthe_phoenix, "~> 2.0"},
You need to have a working Phoenix pubsub configured. Here is what the default looks like if you create a new Phoenix project:
config :my_app, MyAppWeb.Endpoint,
  pubsub_server: :my_pubsub

In your application supervisor add a line after your existing endpoint supervision
line:
    # List all child processes to be supervised
    children = [
      # Start the Ecto repository
      MyAppWeb.Repo,
      # Start the PubSub server
      {Phoenix.PubSub, name: :my_pubsub},
      # Start the endpoint when the application starts
      MyAppWeb.Endpoint,
      {Absinthe.Subscription, MyAppWeb.Endpoint}
    ]

    # See https://hexdocs.pm/elixir/Supervisor.html
    # for other strategies and supported options
    opts = [strategy: :one_for_one, name: MyAppWeb.Supervisor]
    Supervisor.start_link(children, opts)
Note: If your application is deployed in an environment, where the number of CPU cores can differ between the application instances,
be sure to specify a fixed :pool_size option, otherwise the messages will not be delivered reliably between your nodes. This can
happen often on cloud deployment platforms.

{Absinthe.Subscription, pubsub: MyAppWeb.Endpoint, pool_size: 8}
See Absinthe.Subscription.child_spec/1 for more information on the supported
options.
In your MyAppWeb.Endpoint module add:
use Absinthe.Phoenix.Endpoint
Now, you need to configure your socket. I.e. in your MyAppWeb.UserSocket module add:
use Absinthe.Phoenix.Socket,
  schema: MyAppWeb.Schema
Where MyAppWeb.Schema is the name of your Absinthe schema module. See the Absinthe.Phoenix package for more options.

  
    
  
  GraphiQL (optional)


If you're using the GraphiQL plug, in your MyAppWeb.Router, specify the socket option:
forward "/graphiql",
        Absinthe.Plug.GraphiQL,
        schema: MyAppWeb.Schema,
        socket: MyAppWeb.UserSocket
That is all that's required for setup on the server.

  
    
  
  Setting Options


Options like the context can be configured in the connect/2 callback in your
socket module.
defmodule MyAppWeb.UserSocket do
  use Phoenix.Socket
  use Absinthe.Phoenix.Socket,
    schema: MyAppWeb.Schema

  def connect(params, socket) do
    current_user = current_user(params)
    socket = Absinthe.Phoenix.Socket.put_options(socket, context: %{
      current_user: current_user
    })
    {:ok, socket}
  end

  defp current_user(%{"user_id" => id}) do
    MyApp.Repo.get(User, id)
  end

  def id(_socket), do: nil
end

  
    
  
  Schema


Here's an example schema that lets you use subscriptions to get notified when a comment
is submitted to a GitHub repository:
mutation do
  field :submit_comment, :comment do
    arg :repo_name, non_null(:string)
    arg :content, non_null(:string)

    resolve &Github.submit_comment/3
  end
end

subscription do
  field :comment_added, :comment do
    arg :repo_name, non_null(:string)

    # The topic function is used to determine what topic a given subscription
    # cares about based on its arguments. You can think of it as a way to tell the
    # difference between
    # subscription {
    #   commentAdded(repoName: "absinthe-graphql/absinthe") { content }
    # }
    #
    # and
    #
    # subscription {
    #   commentAdded(repoName: "elixir-lang/elixir") { content }
    # }
    #
    # If needed, you can also provide a list of topics:
    #   {:ok, topic: ["absinthe-graphql/absinthe", "elixir-lang/elixir"]}
    config fn args, _ ->
      {:ok, topic: args.repo_name}
    end

    # this tells Absinthe to run any subscriptions with this field every time
    # the :submit_comment mutation happens.
    # It also has a topic function used to find what subscriptions care about
    # this particular comment
    trigger :submit_comment, topic: fn comment ->
      comment.repository_name
    end

    resolve fn comment, _, _ ->
      # this function is often not actually necessary, as the default resolver
      # for subscription functions will just do what we're doing here.
      # The point is, subscription resolvers receive whatever value triggers
      # the subscription, in our case a comment.
      {:ok, comment}
    end

  end
end
Concretely, if client A submits a subscription document:
subscription {
  commentAdded(repoName: "absinthe-graphql/absinthe") {
    content
  }
}
This tells Absinthe to subscribe client A in the :comment_added field on the "absinthe-graphql/absinthe" topic, because that's what comes back from the config function.
Then, if client B submits a mutation:
mutation {
  submitComment(
    repoName: "absinthe-graphql/absinthe"
    content: "Great library!"
  ) {
    id
  }
}
Client B will get the normal response to their mutation, and since they just ask for the id that's what they'll get.
Additionally, the :submit_comment mutation is configured as a trigger on the :comment_added subscription field, so the trigger function is called. That function returns "absinthe-graphql/absinthe" because that's the repository name for the comment, and now Absinthe knows it needs to get all subscriptions on the :comment_added field that have the "absinthe-graphql/absinthe" topic, so client A gets back:
{ "data": { "commentAdded": { "content": "Great library!" } } }
If you want to publish to this subscription manually (not using triggers in the schema) you can do:
Absinthe.Subscription.publish(MyAppWeb.Endpoint, comment, comment_added: "absinthe-graphql/absinthe")
If you want to subscribe to mutations from within your application, you can do:
{:ok, %{"subscribed" => topic}} = Absinthe.run(subscription_query, MyAppWeb.Schema, context: %{pubsub: MyAppWeb.Endpoint})
MyAppWeb.Endpoint.subscribe(topic)

  
    
  
  De-duplicating Updates


By default, Absinthe will resolve each outgoing publish once per individual subscription. This ensures:
	Different GraphQL documents each receive the different fields they requested
	User-specific updates are sent out, in case context contains user-specific data

To improve the scale at which your subscriptions operate, you may tell Absinthe when it is safe to de-duplicate updates. Simply return a context_id from your field's config function:
subscription do
  field :news_article_published, :article do
    config fn _, _ ->
      {:ok, topic: "*", context_id: "global"}
    end
  end
end
Here we return a constant ("global") because our :article type doesn't contain any user-specific fields on it.
Given these three active subscriptions:
# user 1
subscription {
  newsArticlePublished {
    content
  }
}

# user 2
subscription {
  newsArticlePublished {
    content
    author
  }
}

# user 3
subscription {
  newsArticlePublished {
    content
  }
}
Since we provided a context_id, Absinthe will only run two documents per publish to this field:
	Once for user 1 and user 3 because they have the same context ID ("global") and sent the same document.
	Once for user 2. While user 2 has the same context ID ("global"), they provided a different document, so it cannot be de-duplicated with the other two.



  

    
Custom Scalar Types
    

One of the strengths of GraphQL is its extensibility---which doesn't end with
its object types, but is present all the way down to the scalar value level.
Sometimes it makes sense to build custom scalar types to better model your
domain. Here's how to do it.
The GraphQL Specification doesn't define date and datetime types, but Absinthe ships with several pre-built for use via import_types. In this example we'll look at how :datetime is defined.

  
    
  
  Defining a scalar


Supporting additional scalar types is as easy as using the scalar macro and
providing parse and serialize functions.
Here's the definition for :datetime from Absinthe.Type.Custom:
@desc """
The `DateTime` scalar type represents a date and time in the UTC
timezone. The DateTime appears in a JSON response as an ISO8601 formatted
string, including UTC timezone ("Z"). The parsed date and time string will
be converted to UTC and any UTC offset other than 0 will be rejected.
"""
scalar :datetime, name: "DateTime" do
  serialize &DateTime.to_iso8601/1
  parse &parse_datetime/1
end

@spec parse_datetime(Absinthe.Blueprint.Input.String.t) :: {:ok, DateTime.t} | :error
@spec parse_datetime(Absinthe.Blueprint.Input.Null.t) :: {:ok, nil}
defp parse_datetime(%Absinthe.Blueprint.Input.String{value: value}) do
  case DateTime.from_iso8601(value) do
    {:ok, datetime, 0} -> {:ok, datetime}
    {:ok, _datetime, _offset} -> :error
    _error -> :error
  end
end
defp parse_datetime(%Absinthe.Blueprint.Input.Null{}) do
  {:ok, nil}
end
defp parse_datetime(_) do
  :error
end
Scalar definitions, created using the scalar macro from
Absinthe.Schema.Notation, are made up of two major elements:
	A function provided to parse that defines how input is converted
from an AST
value (an Absinthe.Blueprint.Input.t) to a value that's suitable
for use in an argument and passed to a resolver.
	A function provided to serialize that defines how input is
serialized back out when used as a result that should be sent back
to the user.

In this example:
mutation CreatePost {
  post(title: "Second", body: "We're off to a great start!", publishedAt: "2017-11-01T12:00:00Z") {
    id
    publishedAt
  }
}
If our schema defines a :published_at argument with the :datetime type:
field :post, :post do
  arg :published_at, :datetime
  resolve fn _, args, _ ->
    # If the arg is provided, `args.published_at` will be a DateTime struct
    # ...
  end
end
Then, the value of the publishedAt GraphQL field ends up being
parsed by the parse_datetime/1 function; it's been defined as the
parse function for the :datetime type using the parse macro.
The value, "2017-11-01T12:00:00Z", will come into the function as an %Absinthe.Blueprint.Input.String{}, thanks to the hard work of the Absinthe parser and processing pipeline.
The parse_datetime/1 function pulls the string value out of the
input, parses it with DateTime.from_iso8601/1, then returns the
correct result. That result will be used as the value of the
:published_at argument when it's passed to the :post field
resolver.
In the event of an error, the parse function can return either a lone
atom :error or error tuple e.g. {:error, "Reason of error"}. The
reason will be included in the output of the executed GraphQL document.
The serializer of :datetime is a pretty simple affair; it uses the
DateTime.to_iso8601/1 utility function. It would be called to
serialize the %DateTime{} struct for the requested :published_at
field in the result.
More custom scalar examples can be found under Absinthe Wiki - Scalar Recipes.


  

    
Importing Types
    

It doesn't take long for a schema module to become crowded with types,
resolvers, and other customizations.
A good first step in cleaning up your schema is extracting your types,
organizing them into other modules, and then using Absinthe.Schema.Notation.import_types/1
to make them available to your schema.

  
    
  
  Example


Let's say you have a schema that looks something like this:
defmodule MyAppWeb.Schema do
  use Absinthe.Schema

  object :user do
    field :name, :string
  end

  # Rest of the schema...

end
You could extract your :user type into a module, MyAppWeb.Schema.AccountTypes:
defmodule MyAppWeb.Schema.AccountTypes do
  use Absinthe.Schema.Notation

  object :user do
    field :name, :string
  end
end
Note that, unlike your schema module, type modules should use
Absinthe.Schema.Notation, not Absinthe.Schema.

Now, you need to make sure you use import_types to tell your schema
where to find additional types:
defmodule MyAppWeb.Schema do
  use Absinthe.Schema

  import_types MyAppWeb.Schema.AccountTypes

  # Rest of the schema...
end
Important: You should only use import_types from your schema
module; think of it like a manifest.

Now, your schema will be able to resolve any references to your :user type
during compilation.

  
    
  
  What about root types?


Root types (which are defined using the query, mutation, and
subscription macros), can only be defined on the schema module---you
can't extract them, but you can use the import_fields mechanism to
extract their contents.
Here's an example:
query do
  import_fields :account_queries
end
This will look for a matching object type :account_queries, and pull
its fields into the root query type.
For more information, see the guide.


  

    
Importing Fields
    

Sometimes an object type becomes too large and needs to be broken into
pieces. This is especially true of the root query, mutation, and
subscription types that are defined in the schema module itself.
Absinthe provides a mechanism,
Absinthe.Schema.Notation.import_fields/1, to support objects being
able to import fields from other object types.

  
    
  
  An Example


Let's say you're building a content management system. Your root query
type has become unwieldy over time, and your schema looks something
like this:
defmodule MyAppWeb.Schema do
  use Absinthe.Schema

  query do

    @desc "Get all the users, optionally filtering"
    field :users, list_of(:user) do
      # ...
    end

    @desc "Get a user using criteria"
    field :user, :user do
      # ...
    end

    # More account-related fields..

    @desc "Get all the articles, optionally filtering"
    field :articles, list_of(:article) do
      # ...
    end

    @desc "Get an article using criteria"
    field :article, :article do
      # ...
    end

    # More content-related fields...

  end

  # Other types...

end
This could be cleaned up to look something like this:
defmodule MyAppWeb.Schema do
  use Absinthe.Schema

  import_types MyAppWeb.Schema.AccountTypes
  import_types MyAppWeb.Schema.ContentTypes

  query do

    # Using :account_queries from MyAppWeb.Schema.AccountTypes
    import_fields :account_queries

    # Using :content_queries from MyAppWeb.Schema.ContentTypes
    import_fields :content_queries

  end

  # Other types...

end
import_fields here is pulling fields in from separate object types.
Before you can import fields from another object type, make sure
that the type in question is available to your schema. See
the guide on importing types for information
on how that's done.

Here's how those object types are defined.
First, AccountTypes:
defmodule MyAppWeb.Schema.AccountTypes do
  use Absinthe.Schema.Notation

  object :account_queries do

    @desc "Get all the users, optionally filtering"
    field :users, list_of(:user) do
      # ...
    end

    @desc "Get a user using criteria"
    field :user, :user do
      # ...
    end

    # More account-related fields...

  end

  # More account-related types...

end
And ContentTypes:
defmodule MyAppWeb.Schema.ContentTypes do
  use Absinthe.Schema.Notation

  object :content_queries do

    @desc "Get all the articles, optionally filtering"
    field :articles, list_of(:article) do
      # ...
    end

    @desc "Get an article using criteria"
    field :article, :article do
      # ...
    end

    # More content-related fields

  end

  # More content-related types...

end
For more information on import_types, see the guide.


  

    
Using Document Variables
    

GraphQL supports query documents that declare variables that can be accepted to fill-in values. This is a useful mechanism for reusing GraphQL documents---instead of attempting to interpolate values yourself.
	To support variables, simply define them for your query document as the specification expects, and pass in a variables option to Absinthe.run.
	If you're using absinthe_plug, variables are passed in for you automatically after being parsed
from the query parameters or POST body.

Here's an example of defining a non-nullable variable, id, in a document and then executing the document with a value for the variable:
"""
query GetItem($id: ID!) {
  item(id: $id) {
    name
  }
}
"""
|> Absinthe.run(MyAppWeb.Schema, variables: %{"id" => "bar"})

# Result
{:ok, %{data: %{"item" => %{"name" => "Bar"}}}}


  

    
Schema Introspection
    

You can introspect your schema using __schema, __type, and __typename,
as described in the specification.

  
    
  
  Examples


Seeing the names of the types in the schema:
"""
{
  __schema {
    types {
      name
    }
  }
}
""" |> Absinthe.run(MyAppWeb.Schema)
{:ok,
  %{data: %{
    "__schema" => %{
      "types" => [
        %{"name" => "Boolean"},
        %{"name" => "Float"},
        %{"name" => "ID"},
        %{"name" => "Int"},
        %{"name" => "String"},
        ...
      ]
    }
  }}
}
Getting the name of the queried type:
"""
{
  profile {
    name
    __typename
  }
}
""" |> Absinthe.run(MyAppWeb.Schema)
{:ok,
  %{data: %{
    "profile" => %{
      "name" => "Joe",
      "__typename" => "Person"
    }
  }}
}
Getting the name of the fields for a named type:
"""
{
  __type(name: "Person") {
    fields {
      name
      type {
        kind
        name
      }
    }
  }
}
""" |> Absinthe.run(MyAppWeb.Schema)
{:ok,
  %{data: %{
    "__type" => %{
      "fields" => [
        %{
          "name" => "name",
          "type" => %{"kind" => "SCALAR", "name" => "String"}
        },
        %{
          "name" => "age",
          "type" => %{"kind" => "SCALAR", "name" => "Int"}
        },
      ]
    }
  }}
}
Note that you may have to nest several depths of type/ofType, as
type information includes any wrapping layers of List and/or NonNull.

  
    
  
  Configuration


The introspection system can be configured through your application config:
config :absinthe,
  include_deprecated: false  # Defaults to false, set to true for backward compatibility
This configuration affects the default behavior of the includeDeprecated argument in introspection queries. When set to false (the default), deprecated fields and values will not be included in introspection results unless explicitly requested via the includeDeprecated argument. When set to true, deprecated fields and values will be included by default, matching the behavior of previous versions of Absinthe.
Note: Setting include_deprecated to true will break compatibility with the GraphQL specification, which recommends not including deprecated fields by default. Only use this option if you need to maintain compatibility with existing code that expects deprecated fields to be included.
For example, with the default configuration (include_deprecated: false), this query will not include deprecated fields:
{
  __type(name: "User") {
    fields {
      name
    }
  }
}
To include deprecated fields, you must explicitly set the includeDeprecated argument:
{
  __type(name: "User") {
    fields(includeDeprecated: true) {
      name
    }
  }
}
If you set include_deprecated: true in your configuration, the first query would include deprecated fields by default, matching the behavior of previous versions of Absinthe.

  
    
  
  Using GraphiQL


The GraphiQL project is
"an in-browser IDE for exploring GraphQL."
Absinthe provides GraphiQL via a plug in absinthe_plug. See the Plug and Phoenix Guide
for how to install that library. Once installed, usage is simple as:
plug Absinthe.Plug.GraphiQL, schema: MyAppWeb.Schema
If you want to use it at a particular path (in this case graphiql in your Phoenix
router) simply do:
# filename: router.ex
forward "/graphiql", Absinthe.Plug.GraphiQL, schema: MyAppWeb.Schema
This can be trivially reserved to just the :dev elixir environment by doing:
# filename: router.ex
if Mix.env == :dev do
  forward "/graphiql", Absinthe.Plug.GraphiQL, schema: MyAppWeb.Schema
end
If you'd prefer to use a desktop application, we recommend using the pre-built
Electron-based wrapper application,
GraphiQL.app.


  

    
Telemetry
    

Absinthe 1.5 and up uses telemetry to instrument its activity.
Call :telemetry.attach/4 or :telemetry.attach_many/4 to attach your
handler function to any of the following event names:
	[:absinthe, :execute, :operation, :start] when the operation starts
	[:absinthe, :execute, :operation, :stop] when the operation finishes
	[:absinthe, :subscription, :publish, :start] when a subscription starts
	[:absinthe, :subscription, :publish, :stop] when a subscription finishes
	[:absinthe, :resolve, :field, :start] when field resolution starts
	[:absinthe, :resolve, :field, :stop] when field resolution finishes
	[:absinthe, :middleware, :batch, :start] when the batch processing starts
	[:absinthe, :middleware, :batch, :stop] when the batch processing finishes
	[:absinthe, :middleware, :batch, :timeout] whe the batch processing times out

Telemetry handlers are called with measurements and metadata. For details on
what is passed, checkout Absinthe.Phase.Telemetry, Absinthe.Middleware.Telemetry,
and Absinthe.Middleware.Batch.
For async, batch, and dataloader fields, Absinthe sends the final event when
it gets the results. That might be later than when the results are ready. If
you need to know how long the underlying operation took, you'll need to hook
telemetry up to that underlying operation. See, for example, the recommended
telemetry events in the documentation for Ecto.Repo.

  
    
  
  Interactive Telemetry


As an example, you could attach a handler in an iex -S mix shell. Paste in:
:telemetry.attach_many(
  :demo,
  [
    [:absinthe, :resolve, :field, :stop]
  ],
  fn event_name, measurements, metadata, _config ->
    %{
      event_name: event_name,
      measurements: measurements,
      metadata: metadata
    }
    |> IO.inspect()
  end,
  []
)
After a query is executed, you'll see something like:
%{
  event_name: [:absinthe, :resolve, :field, :stop],
  measurements: %{duration: 14000},
  metadata: %{
    id: -576460752303351647,
    middleware: [
      {{Absinthe.Resolution, :call}, &MyApp.Resolvers.resolve_field/3}
    ],
    resolution: :...,
    start_time: 1565830447035742000
  }
}

  
    
  
  Opentelemetry


When using Opentelemetry, one usually wants to correlate spans that are created
in spawned tasks with the main trace. For example, you might have a trace started
in a Phoenix endpoint, and then have spans around database access.
One can correlate manually by attaching the OTel context the task function:
ctx = OpenTelemetry.Ctx.get_current()

Task.async(fn ->
  OpenTelemetry.Ctx.attach(ctx)

  # do stuff that might create spans
end)
When using the async and batch middleware, the tasks are spawned by Absinthe,
so you can't attach the context manually.
Instead, you can add the :opentelemetry_process_propagator package to your
dependencies, which has a Task.async/1 wrapper that will attach the context
automatically. If the package is installed, the middleware will use it in place
of the default Task.async/1.


  

    
Schema Deprecation
    

Use the deprecate option when defining any field or enum value.
	Provide a binary value to give a deprecation reason
	Provide true to just mark it as deprecated

An example:
query do
  field :old_item, :item, deprecate: true
  field :another_old_item, :item, deprecate: "still too old"
end
You can also use the deprecate as a macro inside a block, for instance:
field :age, :integer do
  deprecate
  arg :user_id, non_null(:id)
end
With a reason:
field :ssn, :string do
  deprecate "Privacy concerns"
end
Warning
Warning: Deprecated fields and enum values are not reported by default during introspection.


  

    
Document Adapters
    

Absinthe supports an adapter mechanism that allows developers to define their
schema using one code convention (eg, snake_cased fields and arguments), but
accept query documents and return results (including names in errors) in
another (eg, camelCase). This is useful in allowing both client and server to
use conventions most natural to them.
Absinthe ships with two adapters:
	Absinthe.Adapter.LanguageConventions, which expects schemas to be defined
in snake_case (the standard Elixir convention), translating to/from camelCase
for incoming query documents and outgoing results. This is the default as of v0.3,
and it is highly recommended that it's the adapter you use, as introspection
currently makes certain assumptions about how to return results.
	Absinthe.Adapter.Underscore, which is similar to the LanguageConventions
adapter but converts all incoming identifiers to underscores and does not
modify outgoing identifiers (since those are already expected to be
underscores). Unlike Absinthe.Adapter.Passthrough this does not break
introspection.
	Absinthe.Adapter.Passthrough, which is a no-op adapter and makes no
modifications.

To set the adapter, you provide it as an option to Absinthe.run/3:
Absinthe.run(query, MyAppWeb.Schema,
             adapter: Absinthe.Adapter.TheAdapterName)
Notably, this means you're able to switch adapters on case-by-case basis.
In a Phoenix application, this means you could even support using different
adapters for different clients.
A custom adapter module must merely implement the Absinthe.Adapter behaviour,
in many cases with use Absinthe.Adapter and only overriding the desired
functions.
Note that types that are defined external to your application (including
the introspection types) may not be compatible if you're using a different
adapter.


  

    
Safety Limits
    

A misbehaving client might send a very complex GraphQL query that would require
considerable resources to handle. There are two variations of this problem:
	Complex queries that overwhelm resolution resources.
	Extremely long queries that could take considerable resources to parse. 
(For example, an attacker could craft a long query including thousands of 
undefined fields or directives.)

Either of these could be a vector for a denial-of-service attack in any GraphQL 
service. Absinthe includes mechanisms to protect services each of these.

  
    
  
  Complexity Analysis


To protect against queries that could overwhelm available resources, the
complexity of a query can be estimated before it is resolved and limited to a
specified maximum.
For example, to enable complexity analysis and limit the complexity to a value
of 50 -- if we were using Absinthe.run/3 directly -- we would do this:
Absinthe.run(doc, MyAppWeb.Schema, analyze_complexity: true, max_complexity: 50)
That would translate to the following configuration when using
absinthe_plug (>= v1.2.3):
plug Absinthe.Plug,
  schema: MyAppWeb.Schema,
  analyze_complexity: true,
  max_complexity: 50
The maximum value, 50, is compared to complexity values calculated for each request.

  
    
  
  Complexity Analysis


Here's how the complexity value is calculated:
By default each field in a query will increase the complexity by 1. However, it
can be useful to customize how the complexity value for a field is calculated. This is done in your schema using the
complexity/1 macro, which can accept a function or an explicit integer value.
As an example, when a field is a list, the complexity is often correlated to the
size of the list. To prevent large selections, a field can use a limit argument
with a suitable default (think, for instance, of page sizes during pagination),
and complexity can be calculated keeping that in mind. Here is a schema that
supports analyzing (and limiting) complexity using that approach:
defmodule MyAppWeb.Schema do

  use Absinthe.Schema

  query do
    field :people, list_of(:person) do
      arg :limit, :integer, default_value: 10
      complexity fn %{limit: limit}, child_complexity ->
        # set complexity based on maximum number of items in the list and
        # complexity of a child.
        limit * child_complexity
      end
    end
  end

  object :person do
    field :name, :string
    field :age, :integer
  end

end
For a field, the first argument to the function you supply to complexity/1 is the user arguments
-- just as a field's resolver can use user arguments to resolve its value, the complexity
function that you provide can use the same arguments to calculate the field's complexity.
The second argument passed to your complexity function is the sum of all the complexity scores
of all the fields nested below the current field.
(If a complexity function accepts three arguments, the third will be an
%Absinthe.Resolution{} struct, just as with resolvers.)
If the value of a document's :limit argument was 10 and both name and age were queried for,
the complexity of the :people field would be calculated as 20:
	10, the value of :limit
	times 2, the sum of the complexity of the fields requested on the :person

A field's complexity will default to 1 if it's not set explicitly.
So this would be okay:
{
  people(limit: 10) {
    name
  }
}
But this, at a complexity of 60, wouldn't:
{
  people(limit: 30) {
    name
  }
}

  
    
  
  Complexity limiting


If a document's calculated complexity exceeds the configured limit, resolution
will be skipped and an error will be returned in the result detailing the
calculated and maximum complexities.

  
    
  
  Token Limits


To protect a service from invalid queries that could take considerable resources to parse, 
Absinthe offers the option to configure a maximum limit on tokens in the GraphQL request document. 
If the lexer encounters more tokens than this, it will stop the parse phase and return a phase error 
with the message "Token limit exceeded". This limit is :infinity by default (no limit) 
and can be overridden by providing an integer to the option token_limit to Absinthe.run.


  

    
File Uploads
    

Absinthe provides a unique pattern to support file uploads in conjunction with normal GraphQL field arguments as part of the absinthe_plug package.

  
    
  
  Example


The following schema includes a mutation field that accepts multiple uploaded files as arguments (:users and :metadata):
defmodule MyAppWeb.Schema do
  use Absinthe.Schema

  # Important: Needed to use the `:upload` type
  import_types Absinthe.Plug.Types

  mutation do
    field :upload_file, :string do
      arg :users, non_null(:upload)
      arg :metadata, :upload

      resolve fn args, _ ->
        args.users # this is a `%Plug.Upload{}` struct.

        {:ok, "success"}
      end
    end
  end
end
To send a mutation that includes a file upload, you need to
use the multipart/form-data content type. For example, using cURL:
$ curl -X POST \
-F query="mutation { uploadFile(users: \"users_csv\", metadata: \"metadata_json\")}" \
-F users_csv=@users.csv \
-F metadata_json=@metadata.json \
localhost:4000/graphql

Note how there is a correspondence between the value of the :users argument
and the -F option indicating the associated file.
By treating uploads as regular arguments we get all the usual GraphQL argument
benefits (such as validation and documentation)---which we wouldn't get if
we were merely putting them in the context as in other implementations.

  
    
  
  Integration with Client-side GraphQL Frameworks


	Apollo: apollo-absinthe-upload-link
	Apollo (v1): apollo-absinthe-upload-client (Note: does not support Relay Native as of v1.0.1)
	Relay: (None known. Please submit a pull request updating this information.)



  

    
Testing
    

There are three main approaches to testing GraphQL APIs built with Absinthe:
	Testing resolver functions, since they do most of the work.
	Testing GraphQL document execution directly via Absinthe.run/3, for the bigger picture.
	Outside-in, testing the full HTTP request/response cycle with absinthe_plug.

This guide focuses on the third approach, which we generally recommend since it exercises more
of your application.

  
    
  
  Testing with Absinthe Plug


GraphQL is transport independent, but it's most often served over HTTP. To test HTTP requests with absinthe you'll also need absinthe_plug. This guide will also assume you're using Phoenix, although 
it is possible to use Absinthe without it (see the Plug and Phoenix Setup Guide).

  
    
  
  Example


Say we want to test the following schema:
defmodule MyAppWeb.Schema do
  use Absinthe.Schema

  @fakedb %{
    "1" => %{name: "Bob", email: "bubba@foo.com"},
    "2" => %{name: "Fred", email: "fredmeister@foo.com"}
  }

  query do
    field :user, :user do
      arg :id, non_null(:id)

      resolve &find_user/2
    end
  end

  object :user do
    field :name, :string
    field :email, :string
  end

  defp find_user(%{id: id}, _) do
    {:ok, Map.get(@fakedb, id)}
  end
end
Which we have exposed at the /api endpoint:
defmodule MyAppWeb.Router do
  use Phoenix.Router

  scope "/api" do
    forward "/", Absinthe.Plug, schema: MyAppWeb.Schema
  end
end
The test could look something like this:
defmodule MyAppWeb.SchemaTest do
  use MyAppWeb.ConnCase

  @user_query """
  query getUser($id: ID!) {
    user(id: $id) {
      name
      email
    }
  }
  """

  test "query: user", %{conn: conn} do
    conn =
      post(conn, "/api", %{
        "query" => @user_query,
        "variables" => %{id: 1}
      })

    assert json_response(conn, 200) == %{
             "data" => %{"user" => %{"email" => "bubba@foo.com", "name" => "Bob"}}
           }
  end
end

Phoenix generates the MyAppWeb.ConnCase test helper module. This supplies the
conn variable containing the request and response. It also has helper functions 
such as post/3
and json_response/2.
The query is stored in the @user_query module attribute. We post this document to
the GraphQL endpoint at /api, along with a map of variables which will be 
transformed to arguments for the getUser query.
The response to the query can then be directly asserted to be a JSON object of the right shape.


  

    
Using with JavaScript
    

You can interact with an Absinthe GraphQL server via HTTP (thanks to
absinthe_plug) and websockets (thanks to absinthe_phoenix):
We also have special support for configuring and working with specific
JavaScript frameworks. You can see the guides here:
	Apollo Client
	Relay


  
    
  
  Over HTTP


To integrate a JavaScript application via HTTP, any standard GraphQL
HTTP request (GET/POST) will do.
Here's an example using
isomorphic-fetch:
require('isomorphic-fetch');

fetch('http://localhost:4000/graphql', {
  method: 'POST',
  headers: { 'Content-Type': 'application/json' },
  body: JSON.stringify({ query: '{ posts { title } }' }),
})
  .then(res => res.json())
  .then(res => console.log(res.data));

  
    
  
  Over Websockets


See the @absinthe/socket NPM package
for special support for Absinthe's use of Phoenix channels for GraphQL over websockets, including support for
subscriptions.


  

    
Using with Apollo Client
    

An Apollo client manages its connection to the GraphQL server using links -- which are essentially middleware that tell Apollo how to resolve each query. You can configure Apollo to connect to your Absinthe server via HTTP, websockets, or both.

  
    
  
  Using an HTTP link


Using Apollo with an HTTP link does not require any Absinthe-specific configuration. You can create an HTTP link pointed at your Absinthe server as follows:
import { ApolloClient, InMemoryCache, createHttpLink } from "@apollo/client";

// Create an HTTP link to the Absinthe server.
const link = createHttpLink({
  uri: "http://localhost:4000/graphql"
});

// Apollo also requires you to provide a cache implementation
// for caching query results. The InMemoryCache is suitable
// for most use cases.
const cache = new InMemoryCache();

// Create the client.
const client = new ApolloClient({
  link,
  cache
});
You may find that you need to modify the HTTP request that Apollo makes -- for example, if you wish to send the value of a particular cookie in the Authorization header. The setContext helper allows you to do this, and also demonstrates how links in Apollo can be chained.
import { ApolloClient, InMemoryCache, createHttpLink } from "@apollo/client";
import { setContext } from "@apollo/client/link/context";
import Cookies from "js-cookie";

// Create an HTTP link to the Absinthe server.
const httpLink = createHttpLink({
  uri: "http://localhost:4000/graphql"
});

// Use setContext to create a chainable link object that sets
// the token cookie to the Authorization header.
const authLink = setContext((_, { headers }) => {
  // Get the authentication token from the cookie if it exists.
  const token = Cookies.get("token");

  // Add the new Authorization header.
  return {
    headers: {
      ...headers,
      authorization: token ? `Bearer ${token}` : ""
    }
  };
});

// Chain the HTTP link and the authorization link.
const link = authLink.concat(httpLink);

// Apollo also requires you to provide a cache implementation
// for caching query results. The InMemoryCache is suitable
// for most use cases.
const cache = new InMemoryCache();

// Create the client.
const client = new ApolloClient({
  link,
  cache
});

  
    
  
  Using a websocket link


An HTTP link is suitable for many basic use cases, but if you require two-way communication between the server and the client, you will need to use a websocket link. The most common use case is a client that needs to use GraphQL subscriptions to receive updates from the server when particular events occur. To implement a websocket link, we will need to use the @absinthe/socket and @absinthe/socket-apollo-link packages.
import { ApolloClient, InMemoryCache } from "@apollo/client";
import * as AbsintheSocket from "@absinthe/socket";
import { createAbsintheSocketLink } from "@absinthe/socket-apollo-link";
import { Socket as PhoenixSocket } from "phoenix";
import Cookies from "js-cookie";

// Create a standard Phoenix websocket connection. If you need
// to provide additional params, like an authentication token,
// you can configure them in the `params` option.
const phoenixSocket = new PhoenixSocket("ws://localhost:4000/socket", {
  params: () => {
    if (Cookies.get("token")) {
      return { token: Cookies.get("token") };
    } else {
      return {};
    }
  }
});

// Wrap the Phoenix socket in an AbsintheSocket.
const absintheSocket = AbsintheSocket.create(phoenixSocket);

// Create an Apollo link from the AbsintheSocket instance.
const link = createAbsintheSocketLink(absintheSocket);

// Apollo also requires you to provide a cache implementation
// for caching query results. The InMemoryCache is suitable
// for most use cases.
const cache = new InMemoryCache();

// Create the client.
const client = new ApolloClient({
  link,
  cache
});

  
    
  
  Reconnecting the websocket link


You may find that you periodically need to reconnect the websocket with different parameters. The most common case for this is when a user logs in or logs out; you will then want to refresh their subscriptions to reflect the new authentication state. You can accomplish this by invoking phoenixSocket.conn.close(); from your application code whenever the reconnection needs to happen. Phoenix will notice the closed connection and automatically reconnect. It is important that you provide a function that returns the websocket parameters to the socket params option. If you provide the parameters directly as an object, the new parameters will not be picked up when the websocket reconnects, but if you provide a function, Phoenix invokes the function on each connection to obtain the parameters.
Note that this solution (reconnecting with phoenixSocket.conn.close();) is somewhat unstable because it relies upon an implementation detail of the Phoenix socket. Ideally, a future version of the Phoenix package might add a public API method to reconnect the websocket with new parameters.
import { ApolloClient, InMemoryCache } from "@apollo/client";
import * as AbsintheSocket from "@absinthe/socket";
import { createAbsintheSocketLink } from "@absinthe/socket-apollo-link";
import { Socket as PhoenixSocket } from "phoenix";
import Cookies from "js-cookie";

// Create a standard Phoenix websocket connection. If you need
// to provide additional params, like an authentication token,
// you can configure them in the `params` option.
//
// If you plan to reconnect the socket with updated parameters,
// you must provide a function to the `params` option. If you
// provide the parameters directly as an object, the updated
// parameters will not be picked up when the socket reconnects.
const phoenixSocket = new PhoenixSocket("ws://localhost:4000/socket", {
  params: () => {
    if (Cookies.get("token")) {
      return { token: Cookies.get("token") };
    } else {
      return {};
    }
  }
});

// Wrap the Phoenix socket in an AbsintheSocket.
const absintheSocket = AbsintheSocket.create(phoenixSocket);

// Create an Apollo link from the AbsintheSocket instance.
const link = createAbsintheSocketLink(absintheSocket);

// Apollo also requires you to provide a cache implementation
// for caching query results. The InMemoryCache is suitable
// for most use cases.
const cache = new InMemoryCache();

// Create the client.
const client = new ApolloClient({
  link,
  cache
});

// Later in your application code, when you need to reconnect
// the socket.
phoenixSocket.conn.close();

  
    
  
  Using both HTTP and websocket links


A common configuration for Apollo client applications is to use both HTTP and websocket links -- HTTP for queries and mutations, and a websocket for subscriptions. We can implement this in our client using directional composition with Apollo's split helper.
import {
  ApolloClient,
  InMemoryCache,
  createHttpLink,
  split
} from "@apollo/client";
import { getMainDefinition } from "@apollo/client/utilities";
import { setContext } from "@apollo/client/link/context";
import * as AbsintheSocket from "@absinthe/socket";
import { createAbsintheSocketLink } from "@absinthe/socket-apollo-link";
import { Socket as PhoenixSocket } from "phoenix";
import Cookies from "js-cookie";

// Create an HTTP link to the Absinthe server.
const httpLink = createHttpLink({
  uri: "http://localhost:4000/graphql"
});

// Use setContext to create a chainable link object that sets
// the token cookie to the Authorization header.
const authLink = setContext((_, { headers }) => {
  // Get the authentication token from the cookie if it exists.
  const token = Cookies.get("token");

  // Add the new Authorization header.
  return {
    headers: {
      ...headers,
      authorization: token ? `Bearer ${token}` : ""
    }
  };
});

// Chain the HTTP link and the authorization link.
const authedHttpLink = authLink.concat(httpLink);

// Create a standard Phoenix websocket connection. If you need
// to provide additional params, like an authentication token,
// you can configure them in the `params` option.
const phoenixSocket = new PhoenixSocket("ws://localhost:4000/socket", {
  params: () => {
    if (Cookies.get("token")) {
      return { token: Cookies.get("token") };
    } else {
      return {};
    }
  }
});

// Wrap the Phoenix socket in an AbsintheSocket.
const absintheSocket = AbsintheSocket.create(phoenixSocket);

// Create an Apollo link from the AbsintheSocket instance.
const websocketLink = createAbsintheSocketLink(absintheSocket);

// If the query contains a subscription, send it through the
// websocket link. Otherwise, send it through the HTTP link.
const link = split(
  (operation) => {
    const definition = getMainDefinition(query);

    return (
      definition.kind === "OperationDefinition" &&
      definition.operation === "subscription"
    );
  },
  websocketLink,
  authedHttpLink
);

// Apollo also requires you to provide a cache implementation
// for caching query results. The InMemoryCache is suitable
// for most use cases.
const cache = new InMemoryCache();

// Create the client.
const client = new ApolloClient({
  link,
  cache
});


  

    
Using with Relay
    

While GraphQL specifies what queries, mutations, and object types should look
like, Relay is a client-side implementation of an efficient data storage and
(re-)fetching system that is designed to work with a GraphQL server.
To allow Relay to work its magic on the client side, all GraphQL queries and
mutations need to follow certain conventions. Absinthe.Relay provides
utilities to help you make your server-side schemas Relay-compatible while
requiring only minimal changes to your existing code.
Absinthe.Relay supports three fundamental pieces of the Relay puzzle: nodes,
which are normal GraphQL objects with a unique global ID scheme; mutations,
which in Relay Classic conform to a certain input and output structure; and
connections, which provide enhanced functionality around many-to-one lists
(most notably pagination).

  
    
  
  Using Absinthe.Relay


Make sure you have the absinthe_relay
package configured
as a dependency for your application.
To add Relay support schemas should start with use Absinthe.Relay.Schema, eg:
defmodule Schema do
  use Absinthe.Schema
  use Absinthe.Relay.Schema, :classic

  # ...

end
Note that this schema is being prepared to support Relay Classic; if
you do not provide either a :classic or :modern option,
:classic is currently selected as the default, but a warning is
output; :modern will be the default option in v1.5.

If you're defining your types in a separate type module that you're using via
import_types in your schema, use the Notation module instead:
defmodule Schema.Types do
  use Absinthe.Schema.Notation
  use Absinthe.Relay.Schema.Notation, :classic

  # ...

end
Now you're ready to implement the Relay features you need.

  
    
  
  Nodes


To enable Relay to be clever about caching and (re-)fetching data objects, your
server must assign a globally unique ID to each object before sending it down
the wire. Absinthe will take care of this for you if you provide some additional
information in your schema.
First of all, you must define a :node interface in your schema. Rather than
do this manually, Absinthe.Relay provides a macro so most of the configuration
is handled for you.
Use node interface in your schema:
node interface do
  resolve_type fn
    %YourApp.Model.Person{}, _ ->
      :person
    %YourApp.Model.Business{}, _ ->
      :business
    _, _ ->
      nil
  end
end

# ... mutations, queries ...
For instance, if your query or mutation resolver returns:
{:ok, %YourApp.Model.Business{id: 19, business_name: "ACME Corp.", employee_count: 52}}
Absinthe will pattern-match the value to determine that the object type is
:business. This becomes important when you configure your :business type as a node:
node object :business do  # <-- notice the macro prefix "node"
  field :business_name, non_null(:string)
  field :employee_count, :integer
end
(Note the macro name, node, looks like a prefix to the object macro.)
While it may appear that your :business object type only has two fields,
:business_name and :employee_count, it actually has three. An :id field
is configured for you because you used the node object macro, and because the
:node interface knows how to identify the values returned from your resolvers,
that :id field is automatically set-up to convert internal (in this case,
numeric) IDs to the global ID scheme -- an opaque string (like "UWxf59AcjK=")
will be returned instead.
Important: the global ID is generated based on the object's
unique identifier, which by default is the value of its existing :id
field. This is convenient, because if you are using Ecto, the
primary key :id database field is typically enough to uniquely identify an
object of a given type. It also means, however, that the internal :id of a
node object will not be available to be queried as :id.
	If you wish to generate your global IDs based on something other than the
existing :id field (if, for instance, your internal IDs are returned as _id),
provide the :id_fetcher option (see the documentation).
	If you wish to make your internal ID queryable, you must return it as a
different field (eg, you could define an :internal_id field whose resolver
extracts the raw, internal :id value from the source map/struct).


  
    
  
  Node query field


Ok, so your node objects provide a global :id. How does Relay use it?
Relay expects you to provide a query field called node that accepts a global
ID (as arg :id) and returns the corresponding node object. Absinthe makes it
easy to set this up -- use the node field macro inside your query.
query do
  # ...
  node field do
    resolve fn
      %{type: :person, id: id}, _ ->
        # Get the person from the DB somehow, returning a tuple
        YourApp.Resolver.Person.find(%{id: id}, %{})
      %{type: :business, id: id}, _ ->
        # Get the business from @businesses
        {:ok, Map.get(@businesses, id)}
      # etc.
    end
  end
  # ... more queries ...
end
Notice that the resolver for node field expects the first (args) argument to
contain a :type and :id. These are the node object type identifier and the
internal (non-global) ID, automatically parsed from the global ID. The resolver
looks up the correct value using the internal ID and returns a tuple, as usual.
For more information, see the documentation.

  
    
  
  Converting node IDs to internal IDs for resolvers


If you need to parse a node (global) ID for use in a resolver, there is a
helpful utility, parsing_node_ids/2 that is automatically imported for you.
Here's an example of how it works.
Let's assume we have a field, :employees, that returns a list of :person
objects for a given :business_id -- a node ID:
query do
  field :employees, list_of(:people) do
    arg :business_id, :id
    resolve &resolve_employees/2
  end
end

def resolve_employees(%{business_id: global_id}, _) do
  # But I need an internal ID to look-up the employees!
end
In resolve_employees/2, we could certainly parse out the internal ID manually.
Here's how that would look:
def resolve_employees(%{business_id: global_id}, _) do
  {:ok, %{type: :business, id: internal_id}} =
    Absinthe.Relay.Node.from_global_id(global_id, YourApp.Schema)
  # TODO: find employees using internal_id, return tuple
end
Obviously this can get a bit tedious if we have to do it often. Instead, we can
use parsing_node_ids/2 to wrap our resolver function to do the parsing for
us, invoking our function with the internal ID instead. We just have to tell the
parsing_node_ids/2 what ID field arguments to parse and what the associated
types should be:
query do
  field :employees, list_of(:people) do
    arg :business_id, :id
    resolve parsing_node_ids(&resolve_employees/2, business_id: :business)
  end
end

def resolve_employees(%{business_id: internal_id}, _) do
  # We have an internal ID!
end
This leaves our resolver function virtually unchanged, and keeps our code much
cleaner.

  
    
  
  Mutations


Relay Classic sets some specific constraints around the way arguments and results for
mutations are structured.
In Relay Modern (if you're using the :modern option when defining
the schema), you'll have access to a similar set of macros as
discussed here, but be aware that the constraints mentioned (on
input, on clientMutationId, etc) don't apply. See the
documentation for Absinthe.Relay.Mutation.Notation.Modern for more
specific instructions.

Relay expects mutations to accept exactly one argument, input, an
InputObject. On the JavaScript side, it automatically populates a field on the
input, clientMutationId, and expects to get it back, unchanged, as part of the
result. Thankfully Absinthe.Relay abstracts these details away from the schema
designer, allowing them to focus on any other arguments needed or results
expected.
Important: Remember that input fields (and arguments in
general) cannot be of one of your object types. Use input_object to
model complex argument types.
In this example, we accept a list of multiple :person_input_object values to
insert people into a database.
defmodule YourApp.Schema do
  # ...

  input_object :person_input_object do
    field :first_name, non_null(:string)
    field :last_name, non_null(:string)
    field :age, :integer
  end

  mutation do

    @desc "A mutation that inserts a list of persons into the database"
    payload field :bulk_create_persons do
      input do
        field :persons, list_of(:person_input_object)
      end
      output do
        # fields in the result
      end
      resolve &Resolver.Person.bulk_create/2
    end

    # ... more mutations ...
  end
end
Note the payload macro introduces a Relay mutation, input defines the fields
(inside the input argument), and output defines the fields available as part
of the result.
See the module documentation for more information:
	Absinthe.Relay.Mutation.Notation.Classic
	Absinthe.Relay.Mutation.Notation.Modern


  
    
  
  Referencing existing nodes in mutation inputs


Occasionally, your client may wish to make reference to an existing node in the
mutation input (this happens particularly when manipulating the connection edges
of a parent node).
Incoming IDs for node types may have to be converted to their internal
equivalents so you can persist changes to your backend. For this purpose, you
can use Absinthe.Relay.Node.from_global_id/2 to parse node (global) IDs
manually.
def bulk_create(%{persons: new_persons, group: global_group_id}, _) do
  {:ok, %{type: :group, id: internal_group_id}} = Absinthe.Relay.Node.from_global_id(global_group_id, YourApp.Schema)
  # ... manipulate your DB using internal_group_id
end
If, of course, your client knows the internal IDs (in a peer field to :id, eg,
:internal_id), you can depends on that ID -- but we recommend that you use
node IDs as they are opaque values and it's the more conventional practice.
Important: When using from_global_id, remember to always
match the :type value to ensure the internal ID is for the type you expect,
and a global ID for the wrong type of node hasn't been mistakenly sent to the
server.

  
    
  
  Connections


One of the more popular features of Relay is the rich pagination support provided by its
connections. This medium post
has a good explanation of the full feature set and nomenclature.
For example, you could define a connection for paginating :location objects with:
connection node_type: :location
This will automatically define two new types: :location_connection and :location_edge.
We define a field that uses these types to paginate associated records by using
connection field. Here, for instance, we support paginating a business’s locations:
object :business do
  field :short_name, :string
  connection field :locations, node_type: :location do
    resolve fn
      pagination_args, %{source: business} ->
        Location
        |> where(business_id: ^business.id)
        |> order_by(:inserted_at)
        |> Connection.from_query(&Repo.all/1, pagination_args)
    end
  end
end
We are piping a query for the associated locations into from_query/3 along with the default
relay pagination arguments that allow for pagination. For example, to get just the first 10
locations, use the first argument:
query {
  business(id:"9ea6605e-e6c8-44ea-98d0-1fe6276e193d") {
    shortName
    locations(first:10) {
      edges
        node {
          address1
          city
        }
      }
    }
  }
}
Check the documentation
for more details on connections.
Note: These features do not require using Relay on the client as Apollo
and other client implementations generally support Relay connection configuration.


  

    
Upgrading to v1.4
    

This information is extracted and expanded from the CHANGELOG.

This version included subscriptions, and also came packaged with a number of improvements that required breaking changes.
The breaking changes primarily affect middleware and plugin authors, but some changes (like null handling and changes to error messages) warrant review by all Absinthe users.

  
    
  
  Middleware: Watch Out for Eager Default


Default middleware are now applied eagerly. Although a small change, this will affect anyone who is currently changing the default middleware.

  
    
  
  Before v1.4


Before v1.4, the default middleware was applied "lazily". What this means is if you had a simple field like:
object :user do
  field :name, :string
end
then when it is passed to the middleware/3 callback on a schema, the middleware is an empty list:
def middleware(middleware, %{identifier: :name}, %{identifier: :user}) do
  middleware |> IO.inspect
  #=> []
end
The nice thing about this was that it made it easy to pattern match for the "no middleware supplied" case; you could just match against [].
The problem, however, is that if you wanted to add a simple tracing middleware that runs on every field for example, the "obvious" way to do it seemed like it could be this:
def middleware(middleware, _field, _object) do
  [YourTracer | middleware]
end
... and we just broke our field :name, :string field, and all others like it. No longer did it come back from def middleware as [], and thus our lazy default wasn't applied.
This has tripped up an unreasonable number of people, and it violated the common sense meaning of "default value." It also made it hard for users to determine what default middleware might run on a field. Checking the value passed to middleware/3 didn't help.

  
    
  
  In v1.4


The default middleware is now eager:
def middleware(middleware, %{identifier: :name}, %{identifier: :user}) do
  middleware |> IO.inspect
  #=> [{Absinthe.Middleware.MapGet, :name}]
end
Every field has at least one middleware specified, and the middleware/3 callback has full access to it. Conceptually, it's a lot simpler than the previous approach, in that there isn't some hidden action after the fact that you can't see.
Changing the default is a bit more work now, because you have to explicitly match against the Absinthe default to put something else in its place. However it isn't very difficult (moreover a helper function could be added later to make it even easier).

  
    
  
  Plugins: Change Arguments


Plugins now receive an %Absinthe.Blueprint.Execution{} struct instead of the bare accumulator. This makes it possible for plugins to set or operate on context values. Upgrade your plugins! Change this:
def before_resolution(acc) do
  acc = # doing stuff to the acc here
end
def after_resolution(acc) do
  acc = # doing stuff to the acc here
end
def pipeline(pipeline, acc) do
  case acc do
    # checking on the acc here
  end
end
to:
def before_resolution(%{acc: acc} = exec) do
  acc = # doing stuff to the acc here
  %{exec | acc: acc}
end
def after_resolution(%{acc: acc} = exec) do
  acc = # doing stuff to the acc here
  %{exec | acc: acc}
end
def pipeline(pipeline, exec) do
  case exec.acc do
    # checking on the acc here
  end
end
The reason for this is that you can also access the context within the exec value. When using something like Dataloader, it's important to have easy to the context

  
    
  
  Calling All Resolvers: The Null Literal Has Arrived


Absinthe now supports GraphQL null literals.
null values, when provided as arguments, are passed on to Absinthe resolvers as nil (provided they don't run afoul of a non_null/1 argument constraint).
The concrete effect of this is that your resolvers may need to be updated to take into account that nil is a possible value; in the past arguments would never be passed as nil.
Let's take a look at this example schema snippet.

  
    
  
  Before v1.4


field :avatar_url, :string do
  arg :size, :integer, default_value: 64
  # ... resolver here
end
Before v1.4, you could write a resolver that could always assume a :size would be provided:
resolve fn _, %{size: size}, _ ->
  {:ok, "http://example.com/avatars/test_#{size}x#{size}.png"}
end
If the user doesn't provide a value for :size, it would default to 64:
{
  avatarUrl(size: 32) # => http://example.com/avatars/test_32x32.png
  avatarUrl           # => http://example.com/avatars/test_64x64.png
  # This is invalid in v1.3, as there's no `null` literal:
  avatarUrl(size: null)
}

  
    
  
  In v1.4


In v1.4, users can freely send null, and that value will override any default:
{
  # This didn't change:
  avatarUrl(size: 32)   # => http://example.com/avatars/test_32x32.png
  # Neither did this (default value applied as expected)
  avatarUrl             # => http://example.com/avatars/test_64x64.png
  # But `nil` gets interpolated here, as it overrode the default value of `64`:
  avatarUrl(size: null) # => http://example.com/avatars/test_x.png
}
You may want to change your resolvers to use the default value if nil is received:
resolve fn _, %{size: size}, _ ->
  px = size || 64
  {:ok, "http://example.com/avatars/test_#{px}x#{px}.png"}
end
Or handle it as an explicit match (especially if you want to assign a special semantic value to the now-available null value):
resolve fn
  _, %{size: nil}, _ ->
    {:ok, "http://example.com/avatars/test_default.png"}
  _, %{size: size}, _ ->
    {:ok, "http://example.com/avatars/test_#{size}x#{size}.png"}
end
Some other options to think about:
	Shore up your schema by adding more uses of non_null/1 constraints where it makes sense.
	Add  piece of middleware to strip out nil argument values, universally or focused on a specific set of fields.


  
    
  
  Expect Shorter Error Messages


Errors returned from resolvers no longer say "In field #{field_name}:". The inclusion of the path information obviates the need for this data, and it makes error messages a lot easier to deal with on the frontend.
If you expect certain error messages in your clients, you may need to update your code.


  

    
Upgrading to v1.5
    

This information is extracted and expanded from the CHANGELOG.

This version includes:
	Schema compilation phases
	SDL based schema definitions
	SDL rendering
	Telemetry based instrumentation

Existing macro-based schemas will work as-is, but make sure to note that the schema pipeline executes at compile time.

  
    
  
  Breaking changes



  
    
  
  Default values evaluated at compile time


Default values are evaluated at compile time. For example default_value: DateTime.utc_now() will have its time set at compile time. You probably don't want this :)

  
    
  
  Scalar output validation


Scalar outputs are now type checked and will raise exceptions if the result tries to send the wrong data type in the result.

  
    
  
  Variable types validation


Variable types must now align exactly with the argument type. Previously Absinthe allowed variables of different types to be used by accident as long as the data parsed.

  
    
  
  Field name validation


Added a schema phase to check the validity of field names according to GraphQL spec. Remove the Absinthe.Phase.Schema.Validation.NamesMustBeValid from the schema pipeline if you need to retain the previous behavior.

  
    
  
  Absinthe.Subscription.PubSub


Added node_name/0 callback to Absinthe.Subscription.PubSub behaviour. To retain old behaviour, implement this callback to return Kernel.node/0.

  
    
  
  Absinthe.Traversal


Removed the un-used Absinthe.Traversal module.


  

    
Changelog
    


  
    
  
  1.9.0 (2025-11-21)



  
    
  
  Features


	add GQL sigil to format and lint static GraphQL docs (#1391) (3aef283)


  
    
  
  1.8.0 (2025-11-05)



  
    
  
  Features


	spec: Add support for [@one](https://github.com/one)Of directive (#1386) (01e8e4b)


  
    
  
  Bug Fixes


	typespec: Absinthe.Phase.Subscription.SubscribeSelf.run/2 (#1384) (4230cc4)


  
    
  
  1.7.11 (2025-10-29)



  
    
  
  Bug Fixes


	SDL rendering for directives with input objects (#1375) (02b955c)


  
    
  
  v1.7.10


	Bug Fix: Set include_deprecated default value to true for backwards compatibility
	Bug Fix: Rename to TokenLimitEnforcementTest to fix warning
	Bug Fix: Replace regex with pattern as string to support OTP 28
	Bug Fix: Use Code.ensure_compiled! at compile time


  
    
  
  v1.7.9


	Feature: Allow config/2 to send errors in spec compliant format
	Feature: Add async option to Absinthe.Subscription
	Bug Fix: Avoid table scans on registry
	Bug Fix: Unregsiter duplicate (listening to the same topic) subscriptions individually
	POTENTIALLY BREAKING Feature: Add telemetry event on batch timeout. If you want to keep the behavior from 1.7.8, define a telemetry handler and attach it. For example:

defmodule MyApp.Telemetry do
  require Logger

  def log_absinthe([:absinthe, :middleware, :batch, :timeout], _, metadata, _) do
    Logger.error("Failed to get batching result in #{metadata.timeout}ms for\nfn: #{inspect(metadata.fn)}")
  end
end

# attach

:telemetry.attach("absinthe-batch-timeout", [:absinthe, :middleware, :batch, :timeout], &MyApp.Telemetry.log_absinthe/4, nil)

  
    
  
  1.7.8


	Bugfix: Fixes an issue where schemas would not find their types, or not be found at all.


  
    
  
  1.7.7


	POTENTIALLY BREAKING Bug Fix: Validate variable usage in nested input arguments.This could break incoming documents previously considered valid. Skip the Absinthe.Phase.Document.Arguments.VariableTypesMatch phase to avoid this check. See Absinthe.Pipeline on adjusting the document pipeline.
	#1321 resolves telemetry issues
	Various minor dependency versioning tweaks
	Handle Elixir 1.17 warnings


  
    
  
  1.7.6


	Bugfix: Handle non_null(list_of(:thing)) with null list elements properly
	Bugfix: More non null result handling improvements


  
    
  
  1.7.5


	Feature: Support Dataloader 2.0


  
    
  
  1.7.4


	Bug Fix: Bugfix: multiple pushes per client for subscriptions that have a context_id


  
    
  
  1.7.3


	Bug Fix: OTP 26 and Elixir 1.15 tweaks
	Bug Fix: OTP 25 tweaks
	Bug Fix: Place extra error attributes in error extensions field


  
    
  
  1.7.2


	Bug Fix: Validate type references for invalid wrapped types
	Feature: Add specifiedBy type system directive
	Bug Fix: Object type extensions may be empty
	Bug Fix: Validate input object not being an Enum
	Bug Fix: Deduplicate directives when building schema


  
    
  
  1.7.1


	Breaking Bugfix: Validate repeatable directives on schemas
	Breaking Bugfix: Add "Objects must define fields" schema validation
	Bug Fix: Validate field identifier uniqueness
	Bug Fix: Validate type references for invalid wrapped types
	Bug Fix: Adds optional fix for non compliant built-in scalar Int type. use Absinthe.Schema, use_spec_compliant_int_scalar: true in your schema to use the fixed Int type. It is also advisable to upgrade for custom types if you are leveraging the use of integers outside the GraphQl standard. #1131.
	Feature: Support custom opts in schema pipeline modifiers
	Feature: Support error tuples when scalar parsing fails
	Feature: Convert SDL Language.* structs to SDL notation
	Feature: Support passing the resolution struct to dataloader helper callbacks
	Feature: Add support for type extensions
	Bug Fix: Add type system directives to introspection results
	Bug Fix: Add __private__ field to EnumValueDefinition
	Bug Fix: Fix bug in Schema.absinthe_types(:all) for Persistent Term
	Bug Fix: Fix default enum value check for SDL schema's
	Feature: Add import_directives macro
	Feature: Support type extensions on schema declarations
	Bug Fix: Root objects are marked as referenced correctly
	Bug Fix: Prevent DDOS attacks with long queries
	Feature: pipeline_modifier option to Absinthe.run/3
	Bug Fix: Add end_time_mono to telemetry :stop events


  
    
  
  1.7.0


	POTENTIALLY BREAKING Bug Fix: Validate variable usage in according to spec. This could break incoming documents previously considered valid. Skip the Absinthe.Phase.Document.Arguments.VariableTypesMatch phase to avoid this check. See Absinthe.Pipeline on adjusting the document pipeline.

	Feature: Add GraphQL document formatter

	Bug Fix: Fix Phase.Schema.Validation.InputOutputTypesCorrectlyPlaced not applied to SDL schema's

	Bug Fix: Use inspect/1 to safely encode bad binary samples

	Bug Fix: key :is_type_of not found on Interface 

	Bug Fix: Validate object/interfaces implement all transitive interfaces

	Bug Fix: Fix check unknown types to also cover wrapped types This could break incoming documents previously considered valid. Skip the Absinthe.Phase.Validation.KnownTypeNames phase to avoid this check. See Absinthe.Pipeline on adjusting the document pipeline.

	Bug Fix: Validate field names are unique to an object, interface or an input object



  
    
  
  1.6.7 (Retired)


Originally included the items from 1.7.0, but the spec validation fix was considered
too impactful for a patch release.

  
    
  
  1.6.6


	Feature: Update telemetry dependency to stable ~> 1.0
	Feature: Use makeup_graphql to get GraphQL syntax highlighting in docs
	Bug Fix: Fix exception when field name contains all invalid characters


  
    
  
  1.6.5


	Bug Fix: Fix interface compilation behavior difference between SDL & DSL
	Bug Fix: Allow SDL syntax to contain union without member types
	Bug Fix: Account for prototype schema when rendering SDL via mix task
	Feature: Always inline functions when using persistent_term backend.
	Feature: Support optional open ended scalars


  
    
  
  1.6.4


	Feature: Compress registry tables by default
	Bug Fix: Fix compilation deadlocks on type imports
	Bug Fix: Raise a better error when string serialization fails


  
    
  
  1.6.3


	Bug Fix: Fix unicode bug when encoding parse error


  
    
  
  1.6.2


	Bug Fix: Fix regression in SDL rendering for enum values


  
    
  
  1.6.1


	Feature: Improved serialization failure messages
	Bug Fix: Render null default values in SDL
	Bug Fix: Reduce memory usage of Schema.Manager


  
    
  
  1.6.0


	Feature: Interfaces can now implement Interfaces, matching the latest spec
	Feature: Support for the repeatable directive
	Feature: Enable rendering of Type System Directives in SDL based schemas.
	Feature: Correctly match Introspection type specs
	Bug Fix: Restore dynamic description support (Note: the descriptions are evaluated once --- at compile time)
	Bug Fix: Restore dynamic default_value support (Note: the default_values evaluated once --- at compile time)
	Bug Fix: Restore dynamic Enum value support (Note: the value is evaluated once --- at compile time)
	Bug Fix: Interface nullability corrections
	Bug Fix: Fix field listing for Inputs that import fields
	Bug Fix: Properly trim all descriptions no matter the mechanism used to specify them
	Bug Fix: Fix incorrect specification of __TypeKind
	Bug Fix: Better match introspection schema specification
	Bug Fix: Add missing value to __DirectiveLocation
	Bug Fix: Fix compilation problems with import_types
	Bug Fix: Reduce memory consumption of Subscriptions


  
    
  
  1.5.5


	Bug Fix: Fix for nil in ArgumentsOfCorrectType suggestions


  
    
  
  1.5.4


	Feature: Ensure stable ordering in introspection results.
	Bug Fix: Fix rendering of interfaces in SDL
	Bug Fix: Properly escape single line descriptions in SDL
	Bug Fix: Fix :meta on fields
	Bug Fix: Validate that DirectivesMustBeValid
	Bug Fix: Handle default value rendering with partial field set


  
    
  
  1.5.3


	Bug Fix: Handle null propagation with non_null(list_of(non_null(type))) properly
	Bug Fix: Fix double escaping issue with string literal arguments.


  
    
  
  1.5.2


	Bug Fix: Fix issue with persistent term backend.


  
    
  
  1.5.1


	Bug Fix: Enable hydrating resolve_type on unions. #938
	Bug Fix: #922


  
    
  
  v1.5.0 (Rc)


	Breaking Bug Fix: Variable types must align exactly with the argument type. Previously
Absinthe allowed variables of different types to be used by accident as long as the data parsed.
	Feature (Experimental): :persistent_term based schema backend
	Breaking Change: telemetry event keys changed since the beta release.


  
    
  
  v1.5.0 (Beta)


	Feature: SDL directives, other improvements
	Feature: Output rendered SDL for a schema
	Feature: Substantially lower subscription memory usage.
	Documentation: Testing guide, numerous fixes and updates
	Breaking Change: Scalar outputs are now type checked and will raise exceptions if the result tries to send the wrong data type in the result.
	Breaking Change: telemetry event names changed from the alpha to match an emerging naming convention for tracing.
	Breaking Change: Added phase to check validity of field names according to GraphQL spec. Might break existing schema's. Remove the Absinthe.Phase.Schema.Validation.NamesMustBeValid from the schema pipeline if you want to ignore this.
	Breaking Change: To match the GraphQL spec, we no longer add a non-null error when a resolver on a non-null field explicitly returns its own error.
	Breaking Change: Removed Absinthe.Traversal module


  
    
  
  v1.5.0 (Alpha)


Alpha 0 note: 1.5.0 alpha is safe to use on existing schemas. However, there are no schema validations at the moment, so when writing new ones you may get undefined behaviour if you write an invalid schema.
	COLUMN NUMBERS! The Absinthe Lexer has been rewritten using nimble_parsec and now Absinthe includes column information.
	Complete rewrite of schema internals. This fixes a number of long standing bugs, and provides a variety of new features
	Feature: SDL support
	Feature: Schema decorators
	Feature: Customizable subscription de-duplification. See: https://github.com/absinthe-graphql/absinthe/blob/main/guides/subscriptions.md#de-duplicating-updates
	Feature: Built-in telemetry instrumentation (https://github.com/beam-telemetry/telemetry)
	Breaking Change: default_value: DateTime.utc_now() will have its time set at compile time. IE: DON'T DO THIS. It only worked by accident before anyway, and now it no longer works, which is correct.
	Breaking change: added node_name/0 callback to Absinthe.Subscription.PubSub behaviour. To retain old behaviour, implement this callback to return Kernel.node/0.


  
    
  
  v1.4


For changes pre-v1.5 see the v1.4 branch.


  

    
Absinthe 
    



      
Documentation for the Absinthe package, a toolkit for building GraphQL
APIs with Elixir.
For usage information, see the documentation, which
includes guides, API information for important modules, and links to useful resources.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Absinthe.Formatter - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Formatter 
    



      
Formatter task for graphql
Will format files with the extensions .graphql or .gql

  
    
  
  Example


Absinthe.Formatter.format("{ version }")
"{
  version
}
"
From Elixir 1.13 onwards the Absinthe.Formatter can be added to
the formatter as a plugin:
# .formatter.exs
[
  # Define the desired plugins
  plugins: [Absinthe.Formatter],
  # Remember to update the inputs list to include the new extensions
  inputs: ["{mix,.formatter}.exs", "{config,lib,test}/**/*.{ex,exs}", "{lib,priv}/**/*.{gql,graphql}"]
]

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Absinthe.Lexer - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Lexer 
    




      
        Summary


  
    Functions
  


    
      
        
  
    
    Absinthe.Schema.Manager - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Schema.Manager 
    




      
        Summary


  
    Functions
  


    
      
        
  
    
    Absinthe.Schema.PersistentTerm - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Schema.PersistentTerm 
    



      
Experimental: Persistent Term based Schema Backend
By default, Absinthe schemas are stored in a generated module. If your schema
is called MyAppWeb.Schema, Absinthe creates a MyAppWeb.Schema.Compiled
module containing the structs and other data that Absinthe needs at runtime
to execute GraphQL operations against that schema.
OTP introduced the :persistent_term module to provide many of the same
performance benefits of using compiled modules, without the downsides associated
with manipulating complex structures at compile time.
This module is an experimental effort at using the :persistent_term module
as an Absinthe schema backend. This module has been tested against against
the entire Absinthe test suite and shown to perform perhaps even better
than the compiled module.
To use:
In your schema module:
use Absinthe.Schema
@schema_provider Absinthe.Schema.PersistentTerm
In your application's supervision tree, prior to anywhere where you'd use
the schema:
{Absinthe.Schema, MyAppWeb.Schema}
where MyAppWeb.Schema is the name of your schema.

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Absinthe.Schema.Provider - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Schema.Provider behaviour
    



      
Experimental: Behaviour for providing schema data
This behaviour is experimental and may change significantly in patch releases.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Absinthe.Sigil - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Sigil 
    



      
Provides the ~GQL sigil

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    Absinthe.Resolution.Helpers - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Resolution.Helpers 
    



      
Handy functions for returning async or batched resolution functions
Using Absinthe.Schema.Notation or (by extension) Absinthe.Schema will
automatically import the batch and async helpers. Dataloader helpers
require an explicit import Absinthe.Resolution.Helpers invocation, since
dataloader is an optional dependency.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Absinthe.Schema - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Schema behaviour
    



      
Build GraphQL Schemas

  
    
  
  Custom Schema Manipulation (in progress)


In Absinthe 1.5 and up schemas are built using the same process by which queries are
executed. All the macros in this module and in Notation build up an intermediary tree of structs in the
%Absinthe.Blueprint{} namespace, which we generally call "Blueprint structs".
At the top you've got a %Blueprint{} struct which holds onto some schema
definitions that look a bit like this:
%Blueprint.Schema.SchemaDefinition{
  type_definitions: [
    %Blueprint.Schema.ObjectTypeDefinition{identifier: :query, ...},
    %Blueprint.Schema.ObjectTypeDefinition{identifier: :mutation, ...},
    %Blueprint.Schema.ObjectTypeDefinition{identifier: :user, ...},
    %Blueprint.Schema.EnumTypeDefinition{identifier: :sort_order, ...},
  ]
}
You can see what your schema's blueprint looks like by calling
__absinthe_blueprint__() on any schema or type definition module.
defmodule MyAppWeb.Schema do
  use Absinthe.Schema

  query do

  end
end

> MyAppWeb.Schema.__absinthe_blueprint__()
#=> %Absinthe.Blueprint{...}
These blueprints are manipulated by phases, which validate and ultimately
construct a schema. This pipeline of phases you can hook into like you do for
queries.
defmodule MyAppWeb.Schema do
  use Absinthe.Schema

  @pipeline_modifier MyAppWeb.CustomSchemaPhase

  query do

  end

end

defmodule MyAppWeb.CustomSchemaPhase do
  alias Absinthe.{Phase, Pipeline, Blueprint}

  # Add this module to the pipeline of phases
  # to run on the schema
  def pipeline(pipeline) do
    Pipeline.insert_after(pipeline, Phase.Schema.TypeImports, __MODULE__)
  end

  # Here's the blueprint of the schema, do whatever you want with it.
  def run(blueprint, _) do
    {:ok, blueprint}
  end
end
The blueprint structs are pretty complex, but if you ever want to figure out
how to construct something in blueprints you can always just create the thing
in the normal AST and then look at the output. Let's see what interfaces look
like for example:
defmodule Foo do
  use Absinthe.Schema.Notation

  interface :named do
    field :name, :string
  end
end

Foo.__absinthe_blueprint__() #=> ...

    Absinthe.Schema.Hydrator - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Schema.Hydrator behaviour
    




      
        Summary

    Absinthe.Schema.Notation - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Schema.Notation 
    



      
Provides a set of macros to use when creating a schema. Especially useful
when moving definitions out into a different module than the schema itself.

  
    
  
  Example


defmodule MyAppWeb.Schema.Types do
  use Absinthe.Schema.Notation

  object :item do
    field :id, :id
    field :name, :string
    field :status, :status_enum
  end

  enum :status_enum do
    value :current
    value :discontinued
  end

  # ...

end

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Absinthe.Schema.Prototype - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Schema.Prototype 
    



      
Provides the directives available for SDL schema definitions.
By default, the only directive provided is @deprecated, which supports
a reason argument (of GraphQL type String). This can be used to
mark a field
To add additional schema directives, define your own prototype schema, e.g.:
defmodule MyAppWeb.SchemaPrototype do
  use Absinthe.Schema.Prototype

  directive :feature do
    arg :name, non_null(:string)
    on [:interface]
    # Define `expand`, etc.
  end

  # More directives...
end
Then, set it as the prototype for your schema:
defmodule MyAppWeb.Schema do
  use Absinthe.Schema

  @prototype_schema MyAppWeb.SchemaPrototype

  # Use `import_sdl`, etc...
end

    Absinthe.Type.Argument - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Type.Argument 
    



      
Used to define an argument.
Usually these are defined using Absinthe.Schema.Notation.arg/2

    Absinthe.Type.Custom - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Type.Custom 
    



      
This module contains the following additional data types:
	datetime (UTC)
	naive_datetime
	date
	time
	decimal (only if Decimal is available)

Further description of these types can be found in the source code.
To use: import_types Absinthe.Type.Custom.

      




  

  
    
    Absinthe.Type.Directive - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Type.Directive 
    



      
Used by the GraphQL runtime as a way of modifying execution
behavior.
Type system creators will usually not create these directly.

    Absinthe.Type.Enum - absinthe v1.9.0
    
    

    


  
  

    
Absinthe.Type.Enum 
    



      
Used to define an enum type, a special scalar that can only have a defined set
of values.
See the t type below for details and examples.

  
    
  
  Examples


Given a type defined as the following (see Absinthe.Schema.Notation):
@desc "The selected color channel"
enum :color_channel do
  value :red, as: :r, description: "Color Red"
  value :green, as: :g, description: "Color Green"
  value :blue, as: :b, description: "Color Blue"
  value :alpha, as: :a, deprecate: "We no longer support opacity settings", description: "Alpha Channel"
end
The "ColorChannel" type (referred inside Absinthe as :color_channel) is an
Enum type, with values with names "red", "green", "blue", and "alpha" that map
to internal, raw values :r, :g, :b, and :a. The alpha color channel
is deprecated, just as fields and arguments can be.
You can omit the raw value if you'd like it to be the same as the
identifier. For instance, in this example the value is automatically set to
:red:
enum :color_channel do
  description "The sele